

JOUER  
AVEC  
AMSTRAD

·Stéphane·Laurent·Pascal·

K E R L O C H

SORACOM  
informatique



**JOUEZ**

**AVEC**

**AMSTRAD**



**KERLOCH**

**Stéphane - Laurent - Pascal**

**JOUEZ AVEC AMSTRAD**

**SORACOM Informatique**

**10, Av du G1 de Gaulle - 35170 BRUZ**

---

**«La loi du 11 mars 1957 n'autorisant, aux termes des alinéas 2 et 3 de l'article 41, d'une part que «les copies ou reproductions strictement réservées à l'usage privé du copiste et non destinées à une utilisation collective» et, d'autre part, que les analyses et les courtes citations dans un but d'exemple et d'illustration, «toute représentation ou reproduction intégrale, ou partielle, faite sans le consentement de l'auteur ou de ses ayants-droit ou ayants cause, est illicite» (alinéa premier de l'article 40). Cette représentation ou reproduction, par quelque procédé que ce soit, constituerait donc une contrefaçon sanctionnée par les articles 425 et suivants du Code Pénal.»**

---

# PRÉFACE

Après avoir parcouru le manuel, vous n'avez qu'une seule envie, c'est de découvrir des vrais programmes fonctionnant sur l'AMSTRAD.

Nous vous proposons donc un recueil de programmes aux sujets aussi divers que possible. Vous y trouverez des jeux, des utilitaires ainsi que des programmes utilisant les capacités graphiques et sonores de l'AMSTRAD CPC 464.

Dans ces pages, nous vous offrons notamment un grand jeu d'aventure, "Le Manoir de Rochebrune" qui vous fera frémir comme aux temps des chevaliers.

Vous trouverez également, dans ces programmes, des idées à prendre et transposer dans vos créations personnelles futures.

Tous les programmes que nous vous proposons devront être exécutés dans l'état initial de l'AMSTRAD, c'est-à-dire après avoir tapé :

```
BORDER 1  
PAPER 0  
PEN 1  
INK 1,24  
INK 0,1  
SPEED INK 20,20  
SYMBOL AFTER 1  
SPEED KEY 30,2
```



# DEMONSTRATION COULEUR

Offre spéciale ! 4 programmes pour le prix d'un. En effet, ce programme contient quatre sous-programmes différents qui vous dévoileront les possibilités étonnantes de votre AMSTRAD.

## *FENETRES COLOREES*

Dans la cassette de démonstration, il est question de fenêtres. Qu'est-ce qu'une fenêtre et comment l'utiliser ? Voilà ce que nous vous montrons dans ce petit programme.

## *TABLEAU DES COULEURS*

Dresse l'inventaire des 27 couleurs existant sur l'AMSTRAD.

## *CONVERSATION FUTURISTE*

"Zup, Pluit tut, Buili tut\*", dit le martien.

"Quick, bluzi flupe, Bilibulu, Biout bic, Splop, Tut \*\*", répondit le neptunien.

\*\* "Eh ! On vole à droite dans l'espace, eh ! Chauffard."

\*\* "Eh ! Chauffard toi-même ! Y'a même pas de radar dans ta soucoupe".

Retrouvez nos deux héros dans le troisième sous-programme.

## CUBES COLORES

Ce programme met en évidence les nuances que l'on peut obtenir avec les 27 couleurs, de façon originale.

### QUELQUES DETAILS TECHNIQUES

70-110	Page de présentation
120-190	Sélection du sous-programme
200-300	Fenêtres colorées
310-410	Tableau des couleurs
420-500	Conversation futuriste
510-580	Cubes colorés

```
10 ' *****
20 ' *DEMONSTRATION COULEUR*
30 ' *****
40 '
50 '           par S,L & P Kerloch
60 '
70 MODE 0:BORDER 2:INK 1,18,6:INK 2,15:SPEED INK 20,20:PAPER 0:INK 0,0
80 PEN 1:LOCATE 4,9:PRINT"DEMONSTRATION":LOCATE 7,14:PRINT"COULEUR"
90 PEN 2:FOR A=4 TO 16:LOCATE A,10:PRINT CHR$(208):NEXT A
100 FOR A=7 TO 13:LOCATE A,15:PRINT CHR$(208):NEXT A
110 FOR A=1 TO 4000:NEXT A
120 MODE 1:BORDER 0:INK 1,15:PEN 1:INK 0,0:PAPER 0
130 PRINT:PRINT:PRINT"FENETRES COLOREES"                -> 1"
140 PRINT:PRINT"TABLEAU DES COULEURS"                   -> 2"
150 PRINT:PRINT"CONVERSATION FUTURISTE"                 -> 3"
160 PRINT:PRINT      "CUBES COLORES"                     -> 4"
170 INPUT A
180 ON A GOSUB 200,310,420,510
190 GOTO 120
200 MODE 0
210 FOR a=0 TO 8
220 INK a,a+1
230 NEXT a
```

```

240 FOR a=0 TO 7
250 WINDOW #a,a+1,a+10,a+1,a+12
260 PAPER #a,a
270 CLS #a
280 NEXT a
290 FOR A=0 TO 2000 : NEXT A
300 RETURN
310 MODE 0
320 FOR A=0 TO 27
330 B=A
340 IF B>15 THEN B=B-15
350 INK B,A
360 PEN B
370 PRINT"COULEUR"A
380 PRINT
390 FOR C=1 TO 500:NEXT C
400 NEXT A
410 RETURN
420 MODE 0
430 a=INT(RND(1)*15)+1
440 PEN a
450 B=INT(RND(1)*300)+1:C=INT(RND(1)*300)+1:D=INT(RND(1)*300)+1
460 SOUND 1,B,5,15
470 SOUND 2,C,5,15
480 SOUND 4,D,5,15
490 PRINT CHR$(143):
500 IF INKEY$<>" THEN RETURN ELSE 430
510 MODE 0:INK 0,0:INK 1,1:INK 2,2:INK 3,11:INK 4,9:INK 5,21:INK
  6,22:INK 7,4:INK 8,8:INK 9,17:INK 10,3:INK 11,6:INK 12,16:INK 1
  3,15:INK 14,24:INK 15,25
520 FOR A=0 TO 60 STEP 2: PLOT 100,A+100,1:DRAWR 60,0,1:NEXT A:F
OR B=0 TO 60 STEP 2:PLOT 160,B+100,2:DRAWR 25,25,2:NEXT B:FOR C=
0 TO 25 STEP 2:PLOT 100+C,160+C,3:DRAWR 60,0,3 :NEXT C
530 FOR A=0 TO 60 STEP 2: PLOT 200,A+100,4:DRAWR 60,0,4:NEXT A:F
OR B=0 TO 60 STEP 2:PLOT 260,B+100,5:DRAWR 25,25,5:NEXT B:FOR C=
0 TO 25 STEP 2:PLOT 200+C,160+C,6:DRAWR 60,0,6:NEXT C
540 FOR A=0 TO 60 STEP 2: PLOT 300,A+100,7:DRAWR 60,0,7:NEXT A:F
OR B=0 TO 60 STEP 2:PLOT 360,B+100,8:DRAWR 25,25,8:NEXT B:FOR C=

```

```
0 TO 25 STEP 2:PLOT 300+C,160+C,9:DRAWR 60,0,9:NEXT C
550 FOR A=0 TO 60 STEP 2: PLOT 400.A+100,10:DRAWR 60,0,10:NEXT A
:FOR B=0 TO 60 STEP 2:PLOT 460,B+100,11:DRAWR 25,25,11:NEXT B:FOR
R C=0 TO 25 STEP 2:PLOT 400+C,160+C,12:DRAWR 60,0,12:NEXT C
560 FOR A=0 TO 60 STEP 2: PLOT 500.A+100,10:DRAWR 60,0,13:NEXT A
:FOR B=0 TO 60 STEP 2:PLOT 560,B+100,14:DRAWR 25,25,14:NEXT B:FOR
R C=0 TO 25 STEP 2:PLOT 500+C,160+C,15:DRAWR 60,0,15:NEXT C
570 FOR A=1 TO 3000:NEXT A
580 RETURN
```

# PAPILLONS

Pourquoi "papillons" ? Et bien, tout simplement parce que la disposition des carrés sur l'écran suit le principe des ailes d'un papillon.

Dans ce programme, nous avons sélectionné en mode 20 colonnes, les 16 plus belles couleurs de l'AMSTRAD CPC 464. Il vous apprendra à manier le curseur texte et à gérer les couleurs.



Maintenant, lancez le programme et laissez librement vagabonder votre imagination.

## *QUELQUES DETAILS TECHNIQUES*

70-90	Initialisation des couleurs
100-150	Initialisation des variables
160-240	Affichage
300-370	Présentation

## *TABLEAU DES VARIABLES*

X1	Coordonnée horizontale dans la partie haute de l'écran.
Y1	Coordonnée verticale dans la partie haute de l'écran.
X2	Coordonnée horizontale dans la partie basse de l'écran.
Y2	Coordonnée verticale dans la partie basse de



l'écran.  
Canal de couleur.

```
10 /          *****
20 /          *PAPILLONS*
30 /          *****
40 /
50 /          par S,L & P Kerloch
60 /
70 GOSUB 300
80 MODE 0:INK 0,0:PAPER 0:BORDER 0
90 INK 1,1:INK 2,2:INK 3,4:INK 4,5:INK 5,6:INK 6,8:INK 7,11:INK
8,14:INK 9,15:INK 10,16:INK 11,18:INK 12,19:INK 13,20:INK 14,24:
INK 15,26
100 CLS
110 X1=INT(RND(1)*10)+1
120 Y1=INT(RND(1)*14)+1
130 X2=INT(RND(1)*7)+1
140 Y2=INT(RND(1)*10)+1
150 I=INT(RND(1)*15)+1
160 PEN I
170 LOCATE 11-X1,15-Y1:PRINT CHR$(233)
180 LOCATE 10+X1,15-Y1:PRINT CHR$(233)
190 LOCATE 11-X2,15+Y2:PRINT CHR$(233)
200 LOCATE 10+X2,15+Y2:PRINT CHR$(233)
210 Z=INT(RND(1)*8)+1
220 IF Z<>1 THEN 110
230 IF INKEY$="" THEN 100
240 GOTO 230
300 MODE 0:INK 1,18,6:INK 2,15:INK 0,0:BORDER 2
310 PEN 1
320 LOCATE 6,12:PRINT"PAPILLONS"
330 PEN 2
340 FOR A=6 TO 14:LOCATE A,13:PRINT CHR$(208):NEXT A
350 FOR A=1 TO 6000:NEXT
360 PEN 1:BORDER 0
370 RETURN
```



# TELECRAN

Si vous avez choisi l'AMSTRAD CPC 464 pour ses qualités graphiques, ce programme va vous intéresser. En effet, il s'agit d'un programme de création graphique. Il utilise le mode 20 colonnes en 16 couleurs, mais vous pourrez le modifier pour obtenir des dessins plus précis en 40 ou 80 colonnes.

Pour dessiner, vous utiliserez les quatre flèches ; pour effacer, les flèches + CTRL. La barre d'espace vous permettra de changer d'encre, la touche A sauvera la page écran sur cassette, la touche B chargera un dessin à partir d'une cassette, et la touche C effacera l'écran. Au départ, l'AMSTRAD vous demandera la couleur de l'encre.

Maintenant, à vos pinceaux !!!

## *QUELQUES DÉTAILS TECHNIQUES*

70-150	Initialisation
160-265	Scrutation clavier
270-300	Affichage du point
310-340	Routines d'affichage du curseur
1000-1050	Changement de couleur
2000-2070	Page de présentation
3000-3060	Routine de sauvegarde sur cassette de la page graphique
4000-4070	Routine de chargement de la page graphi-

que à partir de la cassette

### TABLEAU DES VARIABLES

X	Coordonnée horizontale du curseur
Y	Coordonnée verticale du curseur
C	Couleur du curseur
A	Canal de couleur
I\$	Touche frappée au clavier
Q	Canal de couleur (lors des manipulations avec la cassette)

```
10 '          *****
20 '          *TELECRAN*
30 '          *****
40 '
50 '                par S,L & P Kerloch
60 '
70 MODE 0
80 GOSUB 2000
90 c=1
100 DIM COULEUR(15)
110 A=2
120 CLS
130 INK 1,18
140 X=320:Y=200
150 INPUT "",couleur(1):INK 1,couleur(1):LOCATE 1,1:PRINT" "
160 I$=INKEY$
170 IF I$=CHR$(240) THEN GOSUB 330:Y=Y+2:C=A-1:GOTO 270
180 IF I$=CHR$(248) THEN GOSUB 310:Y=Y+2:C=A-1:GOTO 270
190 IF I$=CHR$(241) THEN GOSUB 330:Y=Y-2:C=A-1:GOTO 270
200 IF I$=CHR$(249) THEN GOSUB 310:Y=Y-2:C=A-1:GOTO 270
210 IF I$=CHR$(242) THEN GOSUB 330:X=X-4:C=A-1:GOTO 270
220 IF I$=CHR$(250) THEN GOSUB 310:X=X-4:C=A-1:GOTO 270
230 IF I$=CHR$(243) THEN GOSUB 330:X=X+4:C=A-1:GOTO 270
240 IF I$=CHR$(251) THEN GOSUB 310:X=X+4:C=A-1:GOTO 270
250 IF I$="C" THEN RUN
260 IF I$="A" THEN GOSUB 3000
265 IF I$="B" THEN GOSUB 4000
```

```

270 PLOT X,Y,C
280 PEN 1
290 IF I$=" " THEN GOSUB 1000
300 GOTO 160
310 PLOT X,Y,0
320 RETURN
330 PLOT X,Y,C
340 RETURN
1000 LOCATE 1,1:INPUT",COULEUR(A)
1010 INK A,COULEUR(A)
1020 A=A+1
1030 IF A/16=INT(A/16) THEN A=1
1040 LOCATE 1,1:PRINT" "
1050 RETURN
2000 BORDER 2:INK 0,0:INK 1,18,6:INK 2,15:SPEED INK 20,20
2010 PEN 1
2020 LOCATE 7,12:PRINT"TELECRAN"
2030 PEN 2
2040 FOR A=7 TO 14:LOCATE A,13:PRINT CHR$(208):NEXT A
2050 FOR A=1 TO 6000:NEXT
2060 INK 0,0:PEN 1
2070 BORDER 0:RETURN
3000 SPEED WRITE 1:SAVE "! DESSIN",B,49152,16383
3010 OPENOUT "! COULEUR"
3020 FOR Q=0 TO 15
3030 PRINT #9,COULEUR(Q)
3040 NEXT Q
3050 CLOSEOUT
3060 RETURN
4000 LOAD"!
4010 OPENIN "!
4020 FOR Q=0 TO 15
4030 INPUT #9,COULEUR(Q)
4040 INK Q,COULEUR(Q)
4050 NEXT Q
4060 CLOSEIN
4070 RETURN

```



## COURBES FOLLES

Asseyez-vous dans votre canapé et regardez. C'est beau, non ? Ce programme trace onze courbes différentes en couleur à partir de fonctions polaires paramétriques. Le traçage s'accompagne d'un fond sonore assez insolite puisque la variation du son suit la courbe.

Et ce n'est pas fini ! Quand le traçage de la courbe est terminé, les points défilent à la manière d'un chenillard. Il vous suffira d'appuyer sur une touche du clavier pour passer au dessin suivant. Au départ, l'AMSTRAD vous demandera le numéro de la courbe par laquelle vous voulez commencer. Ça va du simple cercle au magnifique Lissajous en passant par les courbes les plus folles.

Laissez-vous transporter dans le monde fascinant des graphismes et du son (en stéréo, s'il vous plaît !).

*REMARQUE* : Vous ne profiterez de l'effet stéréophonique que si vous possédez un petit casque stéréo (type Walkman).

### *QUELQUES DETAILS TECHNIQUES*

10-100	Initialisation
110-160	Boucle principale
220-700	Routines contenant les formules
710-790	Permutation des couleurs

**TABLEAU DES VARIABLES**

A	Argument (varie de 0° à 360°).
C	Code couleur.
D	Durée de la note.
E	Numéro de la routine.
P	Pas (incrément de A).
S	Longueur de pause.
X	Coordonnée horizontale du point.
Y	Coordonnée verticale du point.

```

1 '          *****
2 '          *COURBES FOLLES*
3 '          *****
4 '
5 '                par S,L & P Kerloch
6 '
10 GOSUB 800
20 MODE 1
30 c=1
40 INK 1,2:INK 2,6:INK 3,24
50 INPUT e
60 IF e>11 THEN 50
70 a=0
80 ORIGIN 320,200
90 EVERY 7,0 GOSUB 690
100 DEG
110 CLS
120 ON e GOSUB 220,260,300,340,380,420,460,500,540,590,640
130 a=a+p
140 SOUND 1,x+200,d,15
150 SOUND 4,y+200,d,15
160 IF a<=360 THEN 120
170 e=e+1:a=0
180 GOTO 710
190 IF e=12 THEN 20

```

```

200 IF INKEY$(<>)" THEN 110 ELSE 200
210 '
220 x=100*COS(a):y=100*SIN(a):p=1
230 PLOT x,y,c
240 d=6:s=12:RETURN
250 '
260 x=100*COS(a)^3:y=100*SIN(a)^3:p=1
270 PLOT x,y,c:DRAW 0,0,c
280 d=8:s=14:RETURN
290 '
300 x=100*COS(a*3)^3:y=100*SIN(a*5)^3:p=0.5
310 PLOT x,y,c
320 d=7:s=14:RETURN
330 '
340 x=100*COS(a*7):y=100*SIN(a*5):p=0.5
350 PLOT x,y,c
360 d=6:s=100:RETURN
370 '
380 x=150*COS(a)^2*SIN(3*a):y=150*SIN(a)^2*COS(3*a):p=0.8
390 PLOT x,y,c
400 d=11:s=13:RETURN
410 '
420 x=40*(COS(5*a)+3)*COS(2*a):y=40*(COS(5*a)+3)*SIN(2*a):p=1
430 PLOT x,y,c
440 d=10:s=20:RETURN
450 '
460 x=40*(COS(11*a)+3)*COS(a):y=40*(COS(11*a)+3)*SIN(a):p=1
470 PLOT x,y,c
480 d=10:s=30:RETURN
490 '
500 x=50*(2*COS(a)+COS(2*a)):y=50*(2*SIN(a)-SIN(2*a)):p=1
510 PLOT x,y,c
520 d=10:s=12:RETURN
530 '
540 x=COS(a)*120+SIN(a*12)*40
550 y=SIN(a)*120+SIN(a*12)*40
560 p=0.5:PLOT x,y,c

```

```

570 d=10:s=70:RETURN
580 '
590 x=50*(2*COS(a)+COS(2*a))*SIN(2*a)-----
600 y=50*(2*SIN(a)-SIN(2*a))*COS(2*a)
610 p=2:PLOT x,y,c
620 d=14:s=30:RETURN
630 '
640 x=-70*(2*SIN(a)-SIN(2*a))*COS(2*a)-----
650 y=70*(2*COS(a)+COS(2*a))*SIN(3*a)
660 p=1.5:PLOT x,y,c
670 d=13:s=30:RETURN
680 '
690 c=c+1:IF c=4 THEN c=1
700 RETURN
710 INK 1,6:INK 2,24:INK 3,2
720 FOR i=0 TO s:NEXT
730 INK 1,24:INK 2,2:INK 3,6
740 FOR i=0 TO s:NEXT
750 INK 1,2:INK 2,6:INK 3,24
760 FOR i=0 TO s
770 I$=INKEY$:IF I$<>" AND E<>12 THEN 110 ELSE IF I$<>" AND E=
12 THEN RUN
780 NEXT
790 GOTO 710
800 MODE 0:INK 1,18,6:INK 2,15:INK 0,0:BORDER 2
810 PEN 1
820 LOCATE 4,12:PRINT"COURBES FOLLES"
830 PEN 2
840 FOR A=4 TO 17:LOCATE A,13:PRINT CHR$(208):NEXT A
850 PEN 1
860 FOR A=1 TO 6000:NEXT A
870 MODE 1:BORDER 0
880 RETURN

```

## TENDRE MELODIE

Un ordinateur qui ne joue pas de musique, n'est pas un ordinateur. Mais comme votre AMSTRAD en est un, nous étions dans l'obligation de vous jouer un petit morceau, en Si bémol, écrit par Guy LACOUR.

Il se joue sur trois canaux en stéréo. Dommage que les DATA soient si longs à taper !



*REMARQUE* : Ne pas mettre le volume trop fort.

### *QUELQUES DETAILS TECHNIQUES*

70-100	Page de présentation
110-160	Lecture des notes et mélodie
170-320	Données

### *TABLEAU DES VARIABLES*

A	Note du canal 1
B	Note du canal 2
C	Note du canal 4



```

10 / *****
20 / *TENDRE MELODIE*
30 / *****
40 /
50 /           par S,L & P Kerloch
60 /
70 MODE 0:BORDER 2:INK 1,18,6:INK 2,15:INK 0,0:PAPER 0:PEN 1:SPE
ED INK 40,40:INK 3,24
80 LOCATE 4,11:PRINT"TENDRE MELODIE"
90 PEN 2:LOCATE 4,12:PRINT STRING$(14,CHR$(208))
100 PEN 3:LOCATE 10,15:PRINT"Guy Lacour":RESTORE
110 READ a,b,c
120 IF a=-1 THEN END
130 SOUND 1,b,40,14
140 SOUND 2,a,40,15
150 SOUND 4,c,40,14
160 GOTO 110
170 DATA 1,159,284,1,159,284,1,134,284,1,134,284,1,106,379,1,106
,379,1,119,379,1,106,379,1,159,478,1,159,478,1,134,478,1,134,478
180 DATA 1,159,478,1,159,478,1,159,478,1,159,478
190 DATA 319,426,190,319,426,190,268,426,213,268,426,213,426
,190,213,426,190,239,426,213
200 DATA 213,426,213,319,426,190,319,426,190,268,426,190,268,426
,190,426,426,159,426,426,159,426,426,159,426,426,159
210 DATA 319,426,190,319,426,190,268,426,213,268,426,213,213,426
,190,213,426,190,239,426,213,213,426,213,319,426,190,319,426,190
,268,426,190,268,426
220 DATA 190,239,426,201,239,426,201,239,426,201,239,426,201
230 DATA 179,268,758,179,268,536,213,319,358,213,319,379,239,358
,268,239,358,358,268,402,319,239,358,358,179,268,379,179,268,536
,134,319,478,134,301
240 DATA 956,142,284,851,142,284,478,142,319,268,142,319,319
250 DATA 150,536,338,150,536,338,213,536,638,213,536,638,239,536
,1,239,536,1,268,536,638,268,536,638
260 DATA 239,536,1,239,536,1,239,536,1136,239,536,1136,239,1,478
,239,213,478,1,239,284,1,268,284,1,213,379,1,213,379,1,213,379,1
,213,379,1,179,402,1,179
270 DATA 402,1
280 DATA 179,402,1,179,402,1,213,402,1,213,402,1,213,402,1,213,4

```

02,1,268,1,1,268,1,1,268,536,1,268,536  
290 DATA 319,536,190,319,536,190,268,536,213,268,536,213,213,478  
,190,213,478,190,239,478,213,213,478,213,319,402,1,319,402,1,319  
,402,536,319,402,536  
300 DATA 268,536,1,268,536,1,268,536,758,268,536,758,319,478,1,3  
19,478,1,319,478,568,319,478,536,319,478,1,319,478,1,319,478,536  
,319,478,536  
310 DATA 159,426,190,159,426,190,159,426,190,159,426,190,159,426  
,190,159,426,190,159,426,190,159,426,190  
320 DATA -1,0,0

## BOITE A RYTHMES

Nous vous proposons maintenant ce petit programme de boîte à rythmes. Il est composé de quatre sous-programmes jouant des sons différents et d'un programme central qui gère tous les bruits. Ce programme principal utilise la commande EVERY qui déclenche une routine à un moment déterminé, en 200<sup>e</sup> de seconde. Il se sert de quatre chronomètres pour déclencher les quatre sous-programmes. C'est une utilisation simple de cette commande très utile (vous pouvez modifier le rythme en changeant le premier paramètre de la commande EVERY).

### *QUELQUES DETAILS TECHNIQUES*

70-90	Page de présentation
100-140	Déclenchement des routines par EVERY
150-170	1 <sup>er</sup> son court
180-200	2 <sup>e</sup> son court
210-240	1 <sup>er</sup> son long
250-290	2 <sup>e</sup> son long

*REMARQUE* : Ce programme ne contient aucune variable.

```

1 ' *****
2 ' *BOITE A RYTHME*
3 ' *****
4 '
5 '           par S,L & P Kerloch
6 '
10 MODE 0:PAPER 0:INK 0,0:INK 1,18,6:INK 2,15:BORDER 2:SPEED INK
    20,20:PEN 1
20 LOCATE 4,12:PRINT"BOITE A RYTHMES"
30 PEN 2:LOCATE 4,13:PRINT STRING$(15,CHR$(208))
40 EVERY 90,0 GOSUB 190
50 EVERY 15,1 GOSUB 120
60 EVERY 30,3 GOSUB 90
70 EVERY 45,2 GOSUB 150
80 IF INKEY$="" THEN 80 ELSE END
90 ENV 1,15,-1,1
100 SOUND 2,200,15,15,1,0,1
110 RETURN
120 ENV 1,15,-1,1
130 SOUND 2,284,15,15,1,0,1
140 RETURN
150 ENV 2,15,-1,3
160 SOUND 1,600,45,15,2,0,10
170 SOUND 4,5,45,15,2,0,1
180 RETURN
190 ENV 3,15,-1,5
200 ENT 1,1,-1,1,2,1,1,1,-1,1
210 SOUND 1,10,45,15,3,1,1
220 SOUND 4,10,45,15,3,1,1
230 RETURN

```

# CLAVIER

Répondez franchement ! N'avez-vous jamais eu envie de jouer d'un instrument ? Nous vous proposons de pianoter sur votre AMSTRAD. Toutes les informations concernant les notes et les touches sont incluses dans le programme. La pression sur une touche seule vous donnera une noire, cette même touche accompagnée de SHIFT donnera une blanche. Vous pourrez modifier l'enveloppe de volume ou de ton dans les lignes 400 et 500 en vous inspirant de l'annexe 1.


*REMARQUE* : Avant de faire RUN, vérifiez bien que vous êtes en minuscules.

## *QUELQUES DETAILS TECHNIQUES*

70-90	Initialisation des couleurs
100-150	Remplissage du fichier de notes
160-210	Scrutation clavier
220	DATA des notes et des touches
300-390	Affichage des informations
400-510	Exécution de la note
600-660	Page de présentation

## *TABLEAU DES VARIABLES*

PERIODE(18)      Tableau de la hauteur du son de la noire



TOUCHE\$(18) Tableau des touches  
D Hauteur du son de la blanche  
I\$ Touche frappée au clavier

```
10 ' *****
20 ' *CLAVIER*
30 ' *****
40 '
50 ' par S,L & P Kerloch
60 '
70 GOSUB 600
80 CLS:MODE 1:INK 1,6:INK 2,24:INK 3,18:INK 0,0
90 GOSUB 300
100 DIM PERIODE(18)
110 DIM TOUCHE$(18)
120 FOR A=0 TO 18
130 READ TOUCHE$(A)
140 READ PERIODE(A)
150 NEXT A
160 I$=INKEY$
170 FOR A=0 TO 18
180 IF I$=TOUCHE$(A) THEN GOSUB 400
190 IF I$=LOWER$(TOUCHE$(A)) THEN D=PERIODE(A):GOSUB 500
200 NEXT A
210 GOTO 160
220 DATA Q,478,W,451,E,426,R,402,T,379,Y,358,U,338,I,318,O,301,P
,284,A,268,S,253,D,239,F,225,G,213,H,201,J,190,K,179,L,169
300 PEN 1:LOCATE 1,6:PRINT STRING$(40,CHR$(208))
310 LOCATE 1,8:PRINT" Q W E R T Y U I O P"
320 PEN 2
330 PRINT:PRINT"DO DO# RE RE# MI FA FA# SOL SOL# LA"
340 PEN 3
350 PRINT:PRINT:PRINT:PRINT" A S D F G H J K L"
360 PEN 2
370 PRINT:PRINT"LA# SI DO DO# RE RE# MI FA FA#"
380 PEN 1:LOCATE 1,18:PRINT STRING$(40,CHR$(208))
```

```
390 RETURN
400 SOUND 2,PERIODE(A),30,15
410 RETURN
500 SOUND 2,D,15,15
510 RETURN
600 MODE 0:INK 0,0:BORDER 2:INK 1,18,6:INK 2,15:SPEED INK 20,20

610 PEN 1
620 LOCATE 7,12:PRINT"CLAVIER"
630 PEN 2
640 LOCATE 7,13:PRINT STRING$( &7,CHR$(208))
650 FOR A=1 TO 4000:NEXT A
660 MODE 1:BORDER 0:RETURN
```

# BOMBARDIER

Mission B 19... Vous devez raser Micropolis avec un bombardier défectueux qui ne lâche que trois bombes à chaque passage, de plus, une panne vous contraint à planer et à perdre de l'altitude. La seule commande encore en service est la barre d'espace pour lâcher un obus. Vous devez détruire les hauts buildings de la ville pour ne pas entrer en collision avec l'un d'eux et vous frayer une piste d'atterrissage.



Si vous mourrez pour la patrie dans cette mission périlleuse, vous resterez un héros dans la mémoire de tous. Adieu, et bonne chance !

## *QUELQUES DETAILS TECHNIQUES*

70-100	Initialisation
110-290	Boucle des passages
300-380	Affichage du score et invitation à rejouer
1000-1050	Remplissage du fichier des buildings
2000-2040	Affichage des buildings
3000-3040	Redéfinition de caractères
4000-4050	Présentation

## *TABLEAU DES VARIABLES*

X	Coordonnée horizontale du bombardier
Y	Coordonnée verticale du bombardier



**BOMBES** Nombre d'obus qu'il est possible de lâcher à  
 chaque passage  
**A** Coordonnée horizontale de l'étage  
**J** Coordonnée verticale de l'étage  
**H** Hauteur de la tour  
**SCORE** Points acquis

```

10 / *****
20 / *BOMBARDIER*
30 / *****

```

```

40 /
50 / par S,L & P Kerloch
60 /

```

```

70 INK 0,0:PAPER 0:INK 1,18,6:PEN 1:INK 2,15:BORDER 2:SPEED INK
20,20

```

```

80 MODE 0:GOSUB 4000:GOSUB 3000:GOSUB 1000:BORDER 1:INK 1,13:INK
2,6:PEN 1:CLS:PAPER 2:GOSUB 2000:PAPER 0

```

```

90 CALL &BD19

```

```

100 INK 3,18:PEN 3:INK 4,10

```

```

110 FOR Y=2 TO 23

```

```

120 BOMBES=3

```

```

130 FOR X=1 TO 19

```

```

140 LOCATE X,Y:PRINT"&"

```

```

150 IF HAUTEUR (X+1,Y)=1 THEN 300

```

```

160 FOR Z=1 TO 200:NEXT Z

```

```

170 IF INKEY$<>" " THEN 260

```

```

180 IF BOMBES<=0 THEN 260 ELSE PEN 4

```

```

190 FOR H=Y TO 24

```

```

200 LOCATE X,H:PRINT""

```

```

210 PEN 3:LOCATE X,Y:PRINT"&":PEN 4

```

```

220 LOCATE X,H:PRINT" "

```

```

230 IF HAUTEUR(X,H+1)=1 THEN HAUTEUR (X,H+1)=0 : LOCATE X,H+1:PR
INT" " :GOTO 250

```

```

240 NEXT H

```

```

250 BOMBES=BOMBES-1:PEN 3

```

```

260 LOCATE X,Y:PRINT" "

```

```

270 SCORE=SCORE+1

```

```

280 NEXT X,Y

```

```

290 SCORE=SCORE+50

```

```

300 CLS
310 INK 1,6,18:INK 2,15
320 PEN 1:LOCATE 5,12:PRINT"SCORE : "SCORE
330 PEN 2
340 FOR A=5 TO 15 :LOCATE A,13:PRINT CHR$(208):NEXT A
350 PRINT:PRINT:PRINT:PRINT:PRINT"VOULEZ-VOUS REJOUER (O/N) "
360 REPONSE$=INKEY$
370 IF REPONSE$="O" OR REPONSE$="o" THEN RUN
380 IF REPONSE$="N" OR REPONSE$="n" THEN END ELSE 360
1000 DIM HAUTEUR(20,25)
1010 FOR A=1 TO 19
1020 H=INT(RND(1)*15)+1
1030 FOR J=25 TO (25-H) STEP -1:HAUTEUR (A,J)=1:NEXT J
1040 NEXT A
1050 RETURN
2000 FOR A=1 TO 20
2010 FOR B=4 TO 25
2020 IF HAUTEUR(A,B)=1 THEN LOCATE A,B:PRINT "%"
2030 NEXT B,A
2040 RETURN
3000 SYMBOL AFTER 36
3010 SYMBOL 37,238,238,238,0,187,187,187,0
3020 SYMBOL 38,0,192,216,252,254,255,24,48
3030 SYMBOL 39,0,28,8,28,28,28,28,8
3040 RETURN
4000 LOCATE 6,12:PRINT"BOMBARDIER"
4010 PEN 2
4020 FOR A=6 TO 15
4030 LOCATE A,13:PRINT CHR$(208):NEXT A
4040 FOR A=1 TO 2000:NEXT A
4050 RETURN

```

## 3D DRIVER

Vous avez déjà sûrement rêvé d'être pilote de Formule 1. Ce rêve va enfin devenir réalité grâce à votre AMSTRAD. Nous n'osons pas comparer notre programme avec le célèbre logiciel "Pôle position" sur ATARI, mais il est bâti selon le même principe. Une route en trois dimensions défile devant votre Formule 1. Vous ne devez pas sortir de la route. Vous utilisez les quatre flèches pour piloter : la flèche droite vous fait aller à droite, la flèche gauche à gauche, la flèche haut vous fait accélérer et la flèche bas vous fait décélérer. L'AMSTRAD calcule votre vitesse et le nombre de kilomètre parcourus. Votre voiture est munie d'une boîte de vitesse automatique. Maintenant, vous êtes sur la grille de départ, le feu passe au vert, et dans un vrombissement d'enfer, vous démarrez sur des routes toujours changeantes.

### *QUELQUES DETAILS TECHNIQUES*

70-140	Initialisation
150-210	Test clavier
220-270	Test de vitesse
280-300	Affichage de la routine et effaçage de l'ancienne route
310-330	Bruit du moteur
340-400	Traçage de la nouvelle route
410-440	Calcul et affichage des kilomètres parcourus

	rus et de la vitesse
450-470	Test de sortie de route
480-550	Redéfinition de caractères
560-650	Explosion
660-720	Page de présentation

### TABLEAU DES VARIABLES

A	Coordonnée horizontale de la voiture
B	Coordonnée verticale de la voiture
C	Bord gauche de la route
L	Largeur de la route
VIT	Vitesse
SOU	Bruit du moteur
A\$	Scrutation clavier
F	Choix de la direction de la route
PT	Kilomètres parcourus

10 ' \*\*\*\*\*

20 ' \*3D DRIVER\*

30 ' \*\*\*\*\*

40 '

50 ' par S,L & P Kerloch

60 '

70 GOSUB 660

80 a=20:b=25:c=250:l=200:vit=10:sou=2000

90 GOSUB 480

100 CLS:MODE 1:INK 1,9:INK 0,0:BORDER 0:INK 3,1:INK 2,6,15:ORIGIN 1,100

110 SPEED INK 2,2

120 FOR r=1 TO 300:PLOT 0,r:DRAW 639,r,3:NEXT r

130 SPEED KEY 10,1

140 ORIGIN 1,1

150 LOCATE a,b:PRINT " "

160 a\$=INKEY\$

170 IF a\$=CHR\$(243) THEN a=a+1

180 IF a\$=CHR\$(242) THEN a=a-1

190 IF A\$=CHR\$(240) THEN VIT=VIT+1:SOU=SOU-30

200 IF A\$=CHR\$(241) THEN VIT=VIT-1:SOU=SOU+30

210 IF a\$=CHR\$(224) THEN END

```

220 IF a<=1 THEN a=1
230 IF a>=38 THEN a=38
240 IF vit>200 THEN vit=200:sou=1500
250 IF vit<0 THEN vit=0:sou=2000
260 IF sou/5>500 THEN sou=1000
270 IF sou/5<=200 THEN sou=1700
280 PEN #1,9:LOCATE a,b:PRINT CHR$(126):CHR$(127)
290 PLOT c,0:DRAW 300,99,0
300 PLOT c+1,0:DRAW 330,99,0
310 SOUND 1,SOU/5,31,15,0,0,1
320 SOUND 2,SOU,31,15,0,0,10
330 SOUND 4,sou/5,31,15
340 f=INT (RND(1)*2)+1
350 IF c<=0 THEN c=c+16:GOTO 390
360 IF c>639-1 THEN c=c-16:GOTO 390
370 IF f=2 THEN c=c+16
380 IF f=1 THEN c=c-16
390 PLOT c,0:DRAW 300,99,2
400 PLOT c+1,0:DRAW 330,99
410 LOCATE 3,1:PRINT "   "
420 pt=pt+vit/1000
430 LOCATE 3,1:PRINT INT(pt):"km "
440 LOCATE 3,3:PRINT VIT:"km/h "
450 IF INT(c/16)>=a-1 THEN GOSUB 560
460 IF INT((c+1)/16)<=a THEN GOSUB 560
470 GOTO 150
480 SYMBOL AFTER 126
490 SYMBOL 126,63,1,1,249,251,255,251,248
500 SYMBOL 127,252,128,128,159,223,255,223,31
510 SYMBOL 128,16,8,81,2,6,10,35
520 SYMBOL 129,18,32,208,32,34,42,208,8
530 SYMBOL 130,0,196,36,0,17,9,201,32
540 SYMBOL 131,12,17,6,128,16,18,4,73
550 RETURN
560 INK 2,0,15:PEN 2
570 INK 1,15,0:INK 2,0,15:INK 3,15,0
580 LOCATE a,b:PRINT CHR$(128):CHR$(129)
590 LOCATE a,b-1:PRINT CHR$(130):CHR$(131)

```

```
600 FOR f=15 TO 8 STEP -1
610 SOUND 6,0,50,f,0,0,20
620 NEXT f
630 FOR d=1 TO 300:NEXT
640 PEN 1
650 RUN
660 MODE 0:INK 1,18,6:INK 2,15:SPEED INK 20,20:INK 0,0:BORDER 2
670 PEN 1
680 LOCATE 6,12:PRINT"3D DRIVER"
690 PEN 2
700 LOCATE 6,13:PRINT STRING$(9,CHR$(208))
710 FOR A=1 TO 4000:NEXT A
720 BORDER 0:PEN 1:MODE 1:RETURN
```

# PAC MAN

C'est une reprise simplifiée sur AMSTRAD du célèbre jeu de gloutons que vous pouvez maintenant trouver sur tous les ordinateurs et les consoles de jeu. Un labyrinthe se dessine sur l'écran, et le jeu commence. Deux fantômes vous poursuivent pour vous croquer alors que vous vous faites un festin de toutes les petites billes que vous rencontrez sur votre passage. Vous vous déplacez à l'aide des quatre flèches.

Bon appétit !

## *QUELQUES DETAILS TECHNIQUES*

70-110	Page de présentation
120-200	Redéfinition
210-260	Affichage des billes
270-500	Affichage du labyrinthe
510-590	Parcours du deuxième fantôme
600-670	Initialisation
680-730	Affichage Pac Man
790-910	Déplacement et affichage des fantômes
920-970	Perte d'une vie
980-1000	Bruit du Pac Man mangeant une bille
1010-1060	Bruit du fantôme dévorant Pac Man
1070-1200	Fin de partie et invitation à rejouer

## TABLEAU DES VARIABLES

X	Coordonnée horizontale du Pac Man
Y	Coordonnée verticale du Pac Man
X1	Coordonnée horizontale du 1 <sup>er</sup> fantôme
Y1	Coordonnée verticale du 1 <sup>er</sup> fantôme
U(300)	Déplacement horizontal du 2 <sup>e</sup> fantôme
V(300)	Déplacement vertical du 2 <sup>e</sup> fantôme
BILLE\$(35,22)	Tableau des billes
PACKMEN	Nombre de vies
I\$	Touche frappée au clavier
SCORE	Nombre de billes avalées
A\$	Caractère représentant le Pac Man

```

10 '          *****
20 '          *PAC MAN*
30 '          *****

```

```

40 '
50 '                par L.Kerloch
60 '

```

```

70 MODE 0:INK 0,0:PAPER 0:BORDER 2:INK 1,6,18:INK 2,15:PEN 1:SPE
ED KEY 5,5:SPEED INK 20,20
80 LOCATE 7,12:PRINT"PAC MAN"
90 PEN 2
100 LOCATE 7,13:PRINT STRING$(7,CHR$(208))
110 FOR X=1 TO 5000:NEXT X
120 SYMBOL AFTER 32
130 SYMBOL 33,62,127,236,248,252,127,62,0
140 SYMBOL 34,124,254,55,31,63,254,124,0
150 SYMBOL 35,56,124,238,234,238,198,68,0
160 SYMBOL 36,68,198,238,254,238,124,56,0
170 SYMBOL 37,0,0,24,60,60,24,0,0
180 SYMBOL 38,28,62,107,127,119,99,127,85
190 SYMBOL 39,255,255,255,255,255,255,255
200 MODE 1:INK 1,0:INK 2,2:INK 3,26:PEN 1 :BORDER 3
210 DIM BILLE$(35,22)
220 FOR X=6 TO 35
230 FOR Y=4 TO 22
240 BILLE$(X,Y)="%"

```



```

,0,-1,0,-1,0,-1,-1,0,-1,0,-1,0,-1,0,0,-1,0,-1,0,-1,0,-1,-1,
0,-1,0,-1,0,-1,0,-1,0,0,1,0,1,0,1,0,1,0,1,0,1,1,0,1,0
550 RESTORE 510
560 DIM U(300),V(300)
570 FOR A=1 TO 200
580 READ U(A),V(A)
590 NEXT A
600 INK 2,2:INK 1,24
610 PACKMEN=3
620 X=25:Y=13:GOSUB 930
630 X1=21:Y1=19
640 X2=9:Y2=11
650 ENV 1,15,-1,4:SOUND 2,478,15,15,1:SOUND 2,358,30,15,1:SOUND
2,478,15,15,1:SOUND 2,358,30,15,1:SOUND 2,319,15,15,1:SOUND 2,28
4,30,15,1:SOUND 2,319,15,15,1:SOUND 2,284,30,15,1:SOUND 2,478,15
,15,1:SOUND 2,358,30,15,1:SOUND 2,478,15,15,1
660 SOUND 2,358,30,15,1:SOUND 2,478,15,15,1:SOUND 2,358,30,15,1
670 FOR WE=1 TO 2000:NEXT WE
680 I$=INKEY$
690 PEN 1
700 IF I$=CHR$(240) AND (TEST(((X-1)*16)+1,INT(400-((Y-1)*16)))=
0) THEN LOCATE X,Y:PRINT " ":Y=Y-1:A$="§":GOTO 740
710 IF I$=CHR$(241) AND (TEST(((X-1)*16)+1,INT(400-((Y+1)*16)))=
0) THEN LOCATE X,Y:PRINT " ":Y=Y+1:A$="§":GOTO 740
720 IF I$=CHR$(242) AND (TEST(((X-2)*16)+1,INT(400-(Y*16)))=0) T
HEN LOCATE X,Y:PRINT " ":X=X-1:A$=CHR$(34):GOTO 740
730 IF I$=CHR$(243) AND (TEST((X*16)+1,INT(400-(Y*16)))=0) THEN
LOCATE X,Y:PRINT " ":X=X+1:A$="!"
740 LOCATE X,Y:PRINT A$
750 IF BILLE$(X,Y)="#" THEN SCORE=SCORE+1:BILLE$(X,Y)="#" :GOSUB
980
760 LOCATE 14,24:PRINT SCORE
770 IF SCORE=166 THEN 1070
780 LOCATE X1,Y1:PRINT BILLE$(X1,Y1)
790 PEN 3
800 IF Y1>Y AND (TEST(((X1-1)*16)+1,INT(400-((Y1-1)*16)))=0) THE
N Y1=Y1-1:GOTO 840
810 IF Y1<Y AND (TEST(((X1-1)*16)+1,INT(400-((Y1+1)*16)))=0) THE
N Y1=Y1+1:GOTO 840

```

```

820 IF X1>X AND (TEST(((X1-2)*16)+1,INT(400-(Y1*16)))=0) THEN X1
=X1-1:GOTO 840
830 IF X1<X AND (TEST((X1*16)+1,INT(400-(Y1*16)))=0) THEN X1=X1+
1
840 LOCATE X1,Y1:PRINT"&"
850 IF X=X1 AND Y=Y1 THEN GOSUB 1010:GOSUB 920
860 PEN 1:LOCATE X2,Y2:PRINT BILLE$(X2,Y2)
870 X2=X2+U(Z):Y2=Y2+V(Z)
880 PEN 3:LOCATE X2,Y2:PRINT "&"
890 IF X=X2 AND Y=Y2 THEN GOSUB 1010:GOSUB 920
900 Z=Z+1:IF Z=201 THEN Z=1
910 GOTO 680
920 PACKMEN=PACKMEN-1:LOCATE X1,Y1:PRINT BILLE$(X1,Y1):X1=21:Y1=
19
930 PEN 1
940 LOCATE 34,24:PRINT PACKMEN
950 PEN 3
960 IF PACKMEN=0 THEN 1070
970 X=25:Y=13:RETURN
980 ENT 1,5,5,1,10,-2,1,10,2,1
990 SOUND 2,300,3,15,0,1
1000 RETURN
1010 ENV 2,15,-1,6
1020 SOUND 6,239,40,15,2
1030 SOUND 6,142,40,15,2
1040 SOUND 6,179,70,15,2
1050 FOR a=1 TO 3000:NEXT a
1060 RETURN
1070 MODE 0:INK 1,18,6:INK 2,15:BORDER 2:INK 0,0:SPEED INK 20,20
:INK 3,24
1080 PEN 1
1090 LOCATE 9,10:PRINT"FIN"
1100 PEN 2
1110 LOCATE 9,11:PRINT STRING$(3,CHR$(208))
1120 PEN 3
1130 LOCATE 3,17:PRINT "Votre score:":SCORE
1140 PEN 1
1150 ENV 1,15,-1,6
1160 SOUND 2,478,60,15,1:SOUND 2,379,60,15,1:SOUND 2,319,60,15,1

```

```
:SOUND 2,239,60,15,1  
1170 FOR A=1 TO 5000:NEXT A  
1180 I$=INKEY$  
1190 IF I$<>" " THEN RUN  
1200 GOTO 1180
```


# STEEPLE CHASE

“...-Alors que Aboudit (en casaque rouge) sort du petit bois, suivi de Topeka (en casaque bleue) et de Ordic (en casaque jaune), qui vient de chuter sur l’obstacle de la sortie du bois... Aboudit et Topeka se disputent la première place, mais Aboudit chute sur l’obstacle des tribunes, et c’est Topeka qui passe la ligne d’arrivée. Il n’y aura pas de photo ; il est évident que le premier est Topeka, déjà vainqueur du Grand Prix d’Automne l’an dernier...”

Vous avez sûrement deviné qu’il s’agit d’un jeu de course hippique. 1 à 3 participants peuvent jouer en même temps. Il faut passer les obstacles en appuyant sur “Z” pour la casaque jaune, sur “,” pour la casaque bleue, et “.” pour la casaque rouge. Il n’y a qu’un gagnant à chaque partie. Cravachez votre cheval car les parieurs ont beaucoup misé sur vous !

## *QUELQUES DETAILS TECHNIQUES*

70-100	Page de présentation
110-280	Initialisation et redéfinition
290-330	Affichage des chevaux
340-370	Test clavier
380-520	Test de rencontre avec un obstacle et fin de la boucle principale.
530-570	Invitation à rejouer



**TABLEAU DES VARIABLES**

NJ    Nombre de joueurs  
O(4)   Coordonnée horizontale des obstacles  
X      Coordonnée horizontale des chevaux  
Y(3)   Coordonnée verticale des chevaux  
I\$     Touche frappée au clavier

```

10 /            *****
20 /            *STEEPLE-CHASE*
30 /            *****
40 /
50 /                        par S,L & P Kerloch
60 /
70 MODE 0:BORDER 2:INK 1,18,6:INK 2,15:SPEED INK 20,20:INK 0,0:P
APER 0:PEN 1
80 LOCATE 4,12:PRINT"STEEPLE-CHASE"
90 PEN 2:LOCATE 4,13:PRINT STRING$(13,CHR$(208))
100 FOR Z=1 TO 3000:NEXT Z
110 MODE 0:BORDER 1:INK 0,1:PAPER #1,0:PEN #1,1
120 SYMBOL AFTER 34
130 SYMBOL 35,&10,&4,&16,&14,&7C,&3A,&21,&42
140 SYMBOL 36,&0,&0,&0,&0,&66,&66,&66,&66
150 SYMBOL 37,&60,&F0,&F0,&F0,&F0,&F0,&F0,&F0
160 SYMBOL 38,&6,&F,&F,&F,&F,&F,&F,&F
170 WINDOW #0,1,20,12,20:INK 14,9
180 INK 15,10:PAPER 15:CLS
190 INK 1,24:INK 2,20:INK 3,6:INK 4,26:INK 5,0:INK 6,2:INK 7,8:I
NK 8,10:INK 9,12:INK 10,14:INK 11,16:INK 12,18:INK 13,22
200 FOR Q=1 TO 4:O(Q)=Q*5:NEXT Q
210 INPUT #1,"NOMBRE DE JOUEURS ? (1-3) ",NJ
220 IF NJ<1 OR NJ>3 THEN 210
230 CLS #1:CLS:GOSUB 600
240 Y(1)=3:Y(2)=5:Y(3)=7
250 FOR A=1 TO 4
260 FOR B=2 TO 8
270 PEN A+9:LOCATE O(A),B:PRINT":"

```

```

280 NEXT B,A
290 FOR X=1 TO 20
300 FOR A=1 TO NJ
310 IF Y(A)=0 THEN 330
320 PEN A:LOCATE X,Y(A):PRINT"##"
330 NEXT A
340 I$=INKEY$
350 IF (I$="Z" OR I$="z") AND Y(1)<>0 THEN Y(1)=2
360 IF I$="," AND NJ>1 AND Y(2)<>0 THEN Y(2)=4
370 IF I$="." AND NJ=3 AND Y(3)<>0 THEN Y(3)=6
380 FOR A=1 TO NJ
390 FOR B=1 TO 4
400 IF Y(A)=0 THEN 420
410 IF X=0(B) AND Y(A)/2<>INT(Y(A)/2) THEN Y(A)=0:SOUND 2,200,30
,15,1
420 NEXT B,A
430 IF Y(1)<>0 THEN Y(1)=3
440 IF Y(2)<>0 THEN Y(2)=5
450 IF Y(3)<>0 THEN Y(3)=7
460 FOR A=1 TO NJ
470 IF Y(A)=0 THEN 510
480 IF X<>0(1) AND X<>0(2) AND X<>0(3) AND X<>0(4) THEN LOCATE X
,Y(A):PRINT" ":GOTO 510
490 PEN (X/5)+9
500 LOCATE X,Y(A):PRINT": "
510 NEXT A
520 NEXT X
530 CLS #2
535 FOR W=100 TO 150:SOUND 2,W,1,15:NEXT W
540 INPUT #2,"Voulez-vous rejouer ?",REP$
550 IF REP$="o" OR REP$="O" THEN RUN
560 IF REP$="n" OR REP$="N" THEN END
570 GOTO 530
600 WINDOW #2,1,20,21,25:PAPER #2,14:CLS #2:PRINT #1,CHR$(22)+CH
R$(1):WINDOW #4,1,20,7,11:PAPER #4,14:CLS #4:PEN #1,1:LOCATE #1,
1,1:PRINT #1,"GRAND PRIX D'AUTOMNE"
610 FOR A=5 TO 9
620 FOR B=4 TO 16

```

```
630 PEN #1,INT(RND(1)*15)+1
640 LOCATE #1,B,A:PRINT #1,"% "
650 PEN #1,INT(RND(1)*15)+1
660 LOCATE #1,B,A:PRINT #1,"& "
670 PEN #1,11
680 LOCATE #1,B,A-1:PRINT #1,"$ "
690 NEXT B,A
700 PEN #1,8:LOCATE #1,4,3:PRINT #1,STRING$(13,CHR$(140))
710 FOR A=4 TO 17 STEP 2
720 PEN #1,8:LOCATE #1,A,3:PRINT #1,CHR$(155)
730 PEN #1,INT(RND(1)*15)+1:LOCATE #1,A,2:PRINT #1,CHR$(136)
740 NEXT A
750 PEN #1,8:LOCATE #1,3,3:PRINT #1,CHR$(136):LOCATE #1,17,3:PRINT #1,CHR$(132)
760 FOR A=4 TO 10
770 LOCATE #1,3,A:PRINT #1,CHR$(138):LOCATE #1,17,A:PRINT #1,CHR$(133)
780 NEXT A
790 LOCATE #1,3,10:PRINT #1,STRING$(15,CHR$(143))
800 PRINT #1,CHR$(22)+CHR$(0)
810 ENV 1,15,-1,2
820 FOR A=1 TO 10:SOUND 2,50,15,15,1:NEXT A
825 FOR A=1 TO 2000:NEXT
830 RETURN
```



# LA MANOIR DE ROCHEBRUNE

Qui n'a jamais rêvé de découvrir le secret du mystérieux Manoir de Rochebrune ?

En effet, dans les villages alentours, on raconte d'étranges légendes à la veillée :

"... Dans des temps reculés, le Seigneur de Rochebrune cacha dans les entrailles de son château trois anneaux ornés de pierres précieuses. Mais pour les réunir, il faut vaincre de terribles monstres et surtout ne pas s'égarer !...

De nombreux aventuriers ont tenté leur chance, mais jamais aucun n'est revenu du terrifiant Manoir de Rochebrune..."

Aujourd'hui, vous avez décidé de pénétrer dans ce lieu maudit afin de retrouver ces trois anneaux de puissance et de richesse.

Que Saint Georges, protecteur des chevaliers, vous garde !

Toutes les touches à utiliser se trouvent sur le pavé numérique.

## DEPLACEMENTS

7 en haut à gauche

8 en haut

9 en haut à droite

- 6 à droite
- 3 en bas à droite
- 2 en bas
- 1 en bas à gauche
- 4 à gauche

## AUTRES COMMANDES

5 ouvrir une porte

0 combattre quand vous vous trouvez près d'un monstre . prendre l'objet sur lequel vous vous trouvez  
"ENTER" inventaire (si vous désirez acheter une fiole ou une clé, il vous suffit alors d'appuyer sur la touche "COPY").

Chaque fiole vous rapporte une vie ; quant aux clés, elles ne peuvent ouvrir que sept portes chacune.  
Chaque combat rapporte 100 écus.

## QUELQUES DETAILS TECHNIQUES

- 100-350 Initialisation et redéfinition
- 360-430 Page de présentation
- 440-680 Fichier du manoir
- 690-1110 Dessin de la salle
- 1120-1410 Choix du contenu de la salle (vide, monstre, clé, anneau ou autre trésor)
- 1420-1600 Scrutation clavier
- 1610-1650 Déplacement du chevalier
- 1660-1750 Combat entre le chevalier et un monstre
- 1760-1780 Affichage de l'objet (trésor, fiole ou clé)
- 1790-1860 Affichage ou déplacement du monstre
- 1870-2190 Ouverture d'une porte
- 2200-2370 Inventaire
- 2380-2490 Achat d'une fiole ou d'une clé
- 2500-2580 Ramassage d'un objet
- 2590-2710 Gloire
- 2720-2840 Mort

## TABLEAU DES VARIABLES

X	Coordonnée horizontale de la salle
Y	Coordonnée verticale de la salle
XX	Coordonnée horizontale du chevalier
YY	Coordonnée verticale du chevalier
CA1\$	Caractère du buste du chevalier
CA2\$	Caractère des jambes du chevalier
NbCles	Nombre de clés
ValCle	Nombre de portes qu'une clé peut encore ouvrir
ECUS	Richesse
VIES	Nombre de vies restant
CASE (15, 15)	Type de salle
CONTENU (15,15)	Type de rencontre
X1	Coordonnée horizontale du monstre
Y1	Coordonnée verticale du monstre
X2	Coordonnée horizontale de son point d'arrivée
Y2	Coordonnée verticale de son point d'arrivée
X3	Coordonnée horizontale de l'objet
Y3	Coordonnée verticale de l'objet
T	Type de trésor
I1	Encre du haut du trésor ou du monstre
I2	Encre du bas du trésor ou du monstre
C1\$	Caractère du haut du trésor ou du monstre
C2\$	Caractère du bas du trésor ou du monstre
I\$	Touche frappée au clavier
XD	Déplacement horizontal du chevalier
YD	Déplacement vertical du chevalier
ANNEAU(1)	Anneau bleu
ANNEAU(2)	Anneau rouge
ANNEAU(3)	Anneau vert

```

10 '          *****
20 '          *   LE   *
30 '          * MANOIR *
40 '          *   DE   *
50 '          * ROCHEBRUNE *
60 '          *****
70 '
80 '                par S,L & P Kerloch
90 '
100 ZY=5:XX=18:YY=11:X=15:Y=11:CA1$=CHR$(213):CA2$=CHR$(214):Nbc
les=0:ValCle=7:ECUS=0:VIES=3:ZZ=0
110 SYMBOL AFTER 212
120 SYMBOL 213,&8,&5C,&5C,&5C,&48,&5C,&5E
130 SYMBOL 214,&FD,&5D,&1D,&14,&24,&45,&22,&10
140 SYMBOL 215,&10,&3A,&3A,&3A,&3A,&12,&3A,&7A
150 SYMBOL 216,&BF,&BA,&B8,&28,&24,&A2,&44,&8
160 SYMBOL 217,&44,&54,&7C,&55,&6D,&39,&12,&7C
170 SYMBOL 218,&B8,&B8,&B9,&BD,&2A,&28,&28,&6C
180 SYMBOL 219,&18,&18,&18,&18,&5A,&DB,&FF,&FF
190 SYMBOL 220,&7E,&3C,&3C,&18,&18,&18,&C,&6
200 SYMBOL 221,&44,&92,&BA,&54,&28,&BA,&92,&BA
210 SYMBOL 222,&7C,&38,&10,&38,&28,&28,&28,&6C
220 SYMBOL 223,&BA,&54,&6C,&38,&10,&54,&FE,&BA
230 SYMBOL 224,&BA,&92,&38,&7C,&7C,&6C,&28,&6C
240 SYMBOL 225,&0,&0,&10,&38,&C,&C,&C,&C
250 SYMBOL 226,&8,&10,&10,&10,&10,&38,&7D,&FE
260 SYMBOL 227,&1C,&14,&1C,&8,&8,&18,&8,&18
270 SYMBOL 228,&10,&38,&10,&38,&6C,&AA,&AA,&7C
280 SYMBOL 229,&EE,&EE,&EE,&0,&BB,&BB,&BB,&0
290 SYMBOL 230,&10,&38,&7C,&54,&7C,&28,&38,&10
300 SYMBOL 231,&39,&7D,&BB,&B9,&B9,&29,&29,&6D
310 SYMBOL 232,&0,&0,&0,&0,&0,&10,&38,&38
320 SYMBOL 233,&10,&10,&10,&38,&7C,&7C,&7C,&38
330 SYMBOL 234,&38,&6C,&44,&44,&6C,&38,&0,&0
340 SYMBOL 235,&0,&14,&41,&18,&BD,&3C,&7E,&FF
350 SYMBOL 236,&7E,&FE,&3D,&BC,&19,&40,&14,&0
360 MODE 0:BORDER 2:INK 1,18,6:INK 2,15:INK 3,6,18:SPEED INK 20,
20:INK 0,0:PAPER 0:PEN 1
370 LOCATE 1,7:PEN 1:PRINT"                Le":PEN 2:PRINT"                ":S

```

```

TRING$(2,CHR$(208)):PRINT
380 PEN 3:PRINT"          MANOIR":PEN 2:PRINT"          ":STRING$(6,CHR$(208)):PRINT
390 PEN 1:PRINT"          de":PEN 2:PRINT"          ":STRING$(2,CHR$(208)):PRINT
400 PEN 3:PRINT"          ROCHEBRUNE":PEN 2:PRINT"          ":STRING$(10,CHR$(208))
410 FOR GW=1 TO 4000:NEXT GW
420 MODE 0:SPEED INK 15,15: BORDER 0
430 INK 0,0:INK 1,2:INK 2,13:INK 3,26:INK 4,6:INK 5,15:INK 6,18:INK 7,24:INK 8,25,26:INK 9,14:INK 10,6,15
440 DIM CASE(15,15)
450 DIM CONTENU(15,15)
460 DATA 5,3,8,3,8,3,8,0,8,3,8,3,7,3,2,2,13,3,4,0,5,3,8,3,7,0,2,1,7,3
470 DATA 9,0,7,1,5,0,13,3,8,3,7,3,9,3,7,3,9,3,7,0,9,3,7,3,6,2,5,3,14,0
480 DATA 5,3,14,3,6,3,9,3,4,2,16,0,8,3,11,3,8,3,11,0,8,0,10,1,9,3,10,3,6,3
490 DATA 6,2,6,3,3,1,1,0,5,3,10,0,1,3,6,3,2,1,10,3,5,0,8,3,8,3,13,0,10,3
500 DATA 16,3,15,3,8,0,14,3,6,3,5,2,11,1,15,1,8,2,4,7,16,3,8,3,4,2,9,3,7,1
510 DATA 16,3,8,3,7,0,3,2,3,0,6,3,6,0,5,0,13,3,8,3,10,0,5,3,7,3,12,5,6,3
520 DATA 6,0,1,1,16,1,8,3,8,2,14,3,9,3,14,0,3,1,2,2,7,3,3,1,16,3,7,2,6,3
530 DATA 9,3,10,3,6,3,2,1,7,3,9,3,8,3,14,0,5,0,8,3,11,3,13,1,10,0,9,3,14,3
540 DATA 5,0,8,3,15,0,13,3,15,3,7,3,1,2,9,3,15,1,8,0,14,3,9,3,13,3,4,2,6,3
550 DATA 6,3,2,1,8,1,10,3,5,0,10,0,6,0,2,1,13,3,4,1,9,0,7,3,9,0,13,3,10,1
560 DATA 9,3,13,1,8,3,13,0,15,0,8,0,11,3,13,3,15,3,8,0,8,3,14,3,5,2,15,3,4,0
570 DATA 5,0,15,3,8,0,10,3,5,1,8,3,10,0,16,3,8,0,4,1,5,3,10,0,4,3,13,3,7,0
580 DATA 6,4,5,3,8,0,7,3,9,3,8,3,8,3,10,0,5,3,8,3,10,1,2,1,8,3,10,0,6,3

```

```

590 DATA 6,3,3,1,1,2,9,0,8,3,7,3,5,1,8,3,10,2,2,2,13,0,8,2,8,3,7
,0,6,3
600 DATA 9,3,8,3,15,3,8,3,8,3,15,3,15,0,8,3,8,0,4,6,9,3,8,3,4,1,
9,3,10,1
610 RESTORE 460
620 FOR A=1 TO 15
630 FOR B=1 TO 15
640 READ CASE(B,A)
650 READ CONTENU(B,A)
660 NEXT B
670 NEXT A
680 CLS
690 X1=0:Y1=0:X2=0:Y2=0:X3=0:Y3=0:C1$="":C2$=""
700 PAPER 4:PEN 2
710 FOR A=1 TO 19
720 LOCATE A,1:PRINT CHR$(229)
730 LOCATE A,2:PRINT CHR$(229)
740 LOCATE A,21:PRINT CHR$(229)
750 LOCATE A,22:PRINT CHR$(229)
760 NEXT A
770 FOR A=1 TO 22
780 LOCATE 1,A:PRINT CHR$(229)
790 LOCATE 19,A:PRINT CHR$(229)
800 NEXT A
810 PAPER 0:IF X=15 AND Y=11 THEN PEN 6:LOCATE 19,11:PRINT CHR$(
199):LOCATE 19,12:PRINT CHR$(199):PEN 5 ELSE PEN 5
820 ON CASE(X,Y) GOSUB 840,860,870,880,890,900,910,920,930,940,9
50,850,960,970,980,990
830 GOTO 1120
840 GOSUB 1030:RETURN
850 RETURN
860 GOSUB 1090:RETURN
870 GOSUB 1000:RETURN
880 GOSUB 1060:RETURN
890 GOSUB 1030:GOSUB 1090:RETURN
900 GOSUB 1000:GOSUB 1030:RETURN
910 GOSUB 1030:GOSUB 1060:RETURN
920 GOSUB 1060:GOSUB 1090:RETURN
930 GOSUB 1000:GOSUB 1090:RETURN

```

```

940 GOSUB 1000:GOSUB 1060:RETURN
950 GOSUB 1000:GOSUB 1030:GOSUB 1060:GOSUB 1090:RETURN
960 GOSUB 1030:GOSUB 1060:GOSUB 1090:RETURN
970 GOSUB 1000:GOSUB 1030:GOSUB 1060:RETURN
980 GOSUB 1000:GOSUB 1060:GOSUB 1090:RETURN
990 GOSUB 1000:GOSUB 1030:GOSUB 1090:RETURN
1000 LOCATE 10,1:PRINT CHR$(210)
1010 LOCATE 10,2:PRINT " "
1020 RETURN
1030 LOCATE 10,22:PRINT CHR$(208)
1040 LOCATE 10,21:PRINT " "
1050 RETURN
1060 LOCATE 1,11:PRINT CHR$(149)
1070 LOCATE 1,12:PRINT CHR$(149)
1080 RETURN
1090 LOCATE 19,11:PRINT CHR$(149)
1100 LOCATE 19,12:PRINT CHR$(149)
1110 RETURN
1120 ON CONTENU(X,Y) GOSUB 1140,1170,1250,1170,1370,1170,1170
1130 GOTO 1420
1140 A=INT(RND(1)*2)
1150 IF A=0 THEN I1=0:I2=7:C1$=CHR$(32):C2$=CHR$(228):T=1
1160 IF A=1 THEN I1=4:I2=6:C1$=CHR$(232):C2$=CHR$(233):T=2
1170 IF CONTENU(X,Y)=2 THEN I1=0:I2=2:C1$=CHR$(32):C2$=CHR$(227)
:T=3
1180 IF CONTENU(X,Y)=4 THEN I1=1:I2=7:C1$=CHR$(232):C2$=CHR$(234)
):T=4
1190 IF CONTENU(X,Y)=6 THEN I1=4:I2=7:C1$=CHR$(232):C2$=CHR$(234)
):T=5
1200 IF CONTENU(X,Y)=7 THEN I1=6:I2=7:C1$=CHR$(232):C2$=CHR$(234)
):T=6
1210 X3=INT(RND(1)*15)+3
1220 Y3=INT(RND(1)*7)+1
1230 Y3=(Y3*2)+3
1240 RETURN
1250 AS=INT(RND(1)*6)+1
1260 IF AS=1 THEN I1=2:I2=11:C1$=CHR$(217):C2$=CHR$(218)
1270 IF AS=2 THEN I1=3:I2=11:C1$=CHR$(219):C2$=CHR$(220)
1280 IF AS=3 THEN I1=4:I2=11:C1$=CHR$(221):C2$=CHR$(222)

```

```

1290 IF AS=4 THEN I1=5:I2=I1:C1$=CHR$(223):C2$=CHR$(224)

1300 IF AS=5 THEN I1=6:I2=I1:C1$=CHR$(225):C2$=CHR$(226)
1310 IF AS=6 THEN I1=8:I2=9:C1$=CHR$(230):C2$=CHR$(231)
1320 X1=INT(RND(1)*15)+3
1330 Y1=INT(RND(1)*7)+1:Y1=(Y1*2)+3
1340 X2=INT(RND(1)*15)+3
1350 Y2=INT(RND(1)*7)+1:Y2=(Y2*2)+3
1360 RETURN
1370 FOR AS=1 TO 6:GOSUB 1260
1380 ZZ=1:GOTO 1420
1390 NEXT AS
1400 I1=10:I2=10:C1$=CHR$(235):C2$=CHR$(236):GOSUB 1320
1410 ZY=1:ZZ=0:SPEED INK 1,2
1420 FOR Z=1 TO ZY:I$=INKEY$
1430 IF ZY=1 THEN GOSUB 1790
1440 PEN 1:LOCATE XX,YY:PRINT CA1$:LOCATE XX,YY+1:PRINT CA2$
1450 IF I$="8" AND YY>4 AND (XX<>X1 OR YY-2<>Y1) THEN YD=-2:XD=
0:GOTO 1610
1460 IF I$="2" AND YY<19 AND (XX<>X1 OR YY+2<>Y1) THEN YD=2:XD=
0:GOTO 1610
1470 IF I$="4" AND XX>2 AND (XX-1<>X1 OR YY<>Y1) THEN XD=-1:YD=
0:CA1$=CHR$(213):CA2$=CHR$(214):GOTO 1610
1480 IF I$="6" AND XX<18 AND (XX+1<>X1 OR YY<>Y1) THEN XD=1:YD=
0:CA1$=CHR$(215):CA2$=CHR$(216):GOTO 1610
1490 IF I$="7" AND YY>4 AND XX>2 AND (XX-1<>X1 OR YY-2<>Y1) THEN
XD=-1:YD=-2:CA1$=CHR$(213):CA2$=CHR$(214):GOTO 1610
1500 IF I$="9" AND YY>4 AND XX<18 AND (XX+1<>X1 OR YY-2<>Y1) THE
N XD=1:YD=-2:CA1$=CHR$(215):CA2$=CHR$(216):GOTO 1610
1510 IF I$="1" AND XX>2 AND YY<19 AND (XX-1<>X1 OR YY+2<>Y1) THE
N XD=-1:YD=2:CA1$=CHR$(213):CA2$=CHR$(214):GOTO 1610
1520 IF I$="3" AND XX<18 AND YY<19 AND (XX+1<>X1 OR YY+2<>Y1) TH
EN YD=2:XD=1:CA1$=CHR$(215):CA2$=CHR$(216):GOTO 1610
1530 IF I$="0" AND (CONTENU(X,Y)=3 OR CONTENU(X,Y)=5) THEN 1660
1540 IF I$="5" AND NbC1es>-1 AND ValC1e>0 THEN 1870
1550 IF I$=CHR$(13) THEN GOSUB 2200
1560 IF I$="." AND XX=X3 AND YY=Y3 THEN 2500
1570 IF CONTENU(X,Y)=1 OR CONTENU(X,Y)=2 OR CONTENU(X,Y)=4 OR CO
NTENU(X,Y)=6 OR CONTENU(X,Y)=7 THEN GOSUB 1760

```

```

1580 NEXT Z
1590 IF CONTENU(X,Y)=3 OR CONTENU(X,Y)=5 THEN GOSUB 1790
1600 GOTO 1420
1610 LOCATE XX,YY:PRINT " ":LOCATE XX,YY+1:PRINT " "
1620 XX=XX+XD:YY=YY+YD
1630 PEN 1
1640 LOCATE XX,YY:PRINT CA1$:LOCATE XX,YY+1:PRINT CA2$
1650 GOTO 1580
1660 IF (X1=XX AND Y1=YY-2) OR (X1=XX-1 AND Y1=YY) OR (X1=XX AND
Y1=YY+2) OR (X1=XX+1 AND Y1=YY) THEN 1680
1670 GOTO 1590
1680 A=INT(RND(1)*2)+1
1690 IF A=1 THEN PEN 4:LOCATE X1,Y1:PRINT CHR$(198):LOCATE X1,Y1
+1:PRINT CHR$(196):CONTENU(X,Y)=0:FOR B=1 TO 200:NEXT B:LOCATE X
1,Y1:PRINT " ":LOCATE X1,Y1+1:PRINT " ":X1=0:Y1=0:ECUS=ECUS+100:I
F ZZ=1 THEN CONTENU(X,Y)=5:GOTO 1390 ELSE GOTO 2590
1700 VIES=VIES-1
1710 PEN 7:LOCATE XX,YY:PRINT CHR$(198):LOCATE XX,YY+1:PRINT CHR
$(196)
1720 IF VIES=0 THEN 2720
1730 FOR B=1 TO 200:NEXT B
1740 LOCATE XX,YY:PRINT " ":LOCATE XX,YY+1:PRINT " "
1750 GOTO 1590
1760 PEN 11:LOCATE X3,Y3:PRINT C1$
1770 PEN 12:LOCATE X3,Y3+1:PRINT C2$
1780 RETURN
1790 LOCATE X1,Y1:PRINT " ":LOCATE X1,Y1+1:PRINT " "
1800 IF X1<X2 THEN X1=X1+1
1810 IF X1>X2 THEN X1=X1-1
1820 IF Y1<Y2 THEN Y1=Y1+2
1830 IF Y1>Y2 THEN Y1=Y1-2
1840 PEN 11:LOCATE X1,Y1:PRINT C1$:PEN 12:LOCATE X1,Y1+1:PRINT C
2$
1850 IF X1=X2 AND Y1=Y2 THEN X2=INT(RND(1)*15)+3:Y2=INT(RND(1)*7
)+1:Y2=(Y2*2)+3
1860 RETURN
1870 IF XX=2 AND YY=11 THEN 1920
1880 IF XX=18 AND YY=11 THEN 1970
1890 IF XX=10 AND YY=3 THEN 2020

```

```

1900 IF XX=10 AND YY=19 THEN 2070
1910 GOTO 1580
1920 IF CASE(X,Y)=4 OR CASE(X,Y)=7 OR CASE(X,Y)=8 OR CASE(X,Y)=1
0 OR CASE(X,Y)=11 OR CASE(X,Y)=13 OR CASE(X,Y)=14 OR CASE(X,Y)=1
5 THEN 1940
1930 GOTO 1580
1940 GOSUB 2160:GOSUB 2120
1950 X=X-1:XX=18
1960 GOTO 2110
1970 IF CASE(X,Y)=2 OR CASE(X,Y)=5 OR CASE(X,Y)=8 OR CASE(X,Y)=9
OR CASE(X,Y)=11 OR CASE(X,Y)=13 OR CASE(X,Y)=15 OR CASE(X,Y)=16
THEN 1990
1980 GOTO 1580
1990 GOSUB 2160:GOSUB 2120
2000 X=X+1:XX=2
2010 GOTO 2110
2020 IF CASE(X,Y)=3 OR CASE(X,Y)=6 OR CASE(X,Y)=9 OR CASE(X,Y)=1
0 OR CASE(X,Y)=11 OR CASE(X,Y)=14 OR CASE(X,Y)=15 OR CASE(X,Y)=1
6 THEN 2040
2030 GOTO 1580
2040 GOSUB 2160:GOSUB 2120
2050 Y=Y-1:YY=19
2060 GOTO 2110
2070 IF CASE(X,Y)=1 OR CASE(X,Y)=5 OR CASE(X,Y)=6 OR CASE(X,Y)=7
OR CASE(X,Y)=11 OR CASE(X,Y)=13 OR CASE(X,Y)=14 OR CASE(X,Y)=16
THEN 2090
2080 GOTO 1580
2090 GOSUB 2160:GOSUB 2120
2100 Y=Y+1:YY=3
2110 GOTO 690
2120 IF CONTENU(X,Y)=1 OR CONTENU(X,Y)=2 THEN LOCATE X3,Y3:PRINT
" ":LOCATE X3,Y3+1:PRINT " "
2130 IF CONTENU(X,Y)=3 THEN LOCATE X1,Y1:PRINT " ":LOCATE X1,Y1+
1:PRINT " "
2140 LOCATE XX,YY:PRINT " ":LOCATE XX,YY+1:PRINT " "
2150 RETURN
2160 ValCle=ValCle-1
2170 IF ValCle=0 THEN NbCles=NbCles-1:ValCle=7
2180 IF NbCles=-1 THEN GOSUB 2200:IF QW=2000 THEN END

```

```

2190 RETURN
2200 FOR QW=3 TO 20
2210 LOCATE 2,QW:PRINT"          " ' 17 ESPACES
2220 NEXT QW
2230 PEN 4:LOCATE 5,4:PRINT"POSSESSIONS"
2240 LOCATE 3,7:PRINT CHR$(232)
2250 PEN 6:LOCATE 3,8:PRINT CHR$(233)::PEN 5:PRINT" VIES ":"VIE
S
2260 PEN 2:LOCATE 3,12:PRINT CHR$(227)::PEN 5:PRINT" CLES ":"Nb
Cles+1
2270 PEN 7:LOCATE 3,16:PRINT CHR$(228)::PEN 5:PRINT" ECUS ":"EC
US
2280 IF ANNEAU(1)=1 THEN PEN 7:LOCATE 8,20:PRINT CHR$(234):PEN 1
:LOCATE 8,19:PRINT CHR$(232)
2290 IF ANNEAU(2)=1 THEN PEN 7:LOCATE 10,20:PRINT CHR$(234):PEN
4:LOCATE 10,19:PRINT CHR$(232)
2300 IF ANNEAU(3)=1 THEN PEN 7:LOCATE 12,20:PRINT CHR$(234):PEN
6:LOCATE 12,19:PRINT CHR$(232)
2310 FOR QW=1 TO 2000
2320 IF INKEY#=CHR$(224) THEN 2380
2330 NEXT QW
2340 FOR QW=3 TO 20
2350 LOCATE 2,QW:PRINT"          " ' 17 ESPACES
2360 NEXT QW
2370 RETURN
2380 FOR QW=3 TO 20:LOCATE 2,QW:PRINT"          " ' 17 ES
PACES
2390 NEXT QW:PEN 4:LOCATE 7,4:PRINT"MAGASIN"
2400 PEN 1:LOCATE 3,7:PRINT"Votre bien":ECUS
2410 PEN 4:LOCATE 3,9:PRINT CHR$(232):PEN 6:LOCATE 3,10:PRINT CH
R$(233)::PEN 5:PRINT" FIOLE 150 (1)
2420 PEN 2:LOCATE 3,13:PRINT CHR$(227)::PEN 5:PRINT" CLE 100 (
2)
2430 PEN 1:LOCATE 3,16:INPUT "Achat ":"A$:A=VAL(A$)
2440 IF A=1 AND ECUS>=150 THEN 2470
2450 IF A=2 AND ECUS>=100 THEN ECUS=ECUS-100:GOTO 2480
2460 GOTO 2330
2470 ECUS=ECUS-150:VIES=VIES+1:GOTO 2340
2480 IF ValCle=0 THEN ValCle=7 ELSE NbCles=NbCles+1

```

```

2490 GOTO 2340
2500 IF T=3 THEN NbCles=NbCles+1
2510 IF T=1 THEN ECUS=ECUS+200
2520 IF T=2 THEN VIES=VIES+1
2530 IF T=4 THEN ANNEAU(1)=1:ECUS=ECUS+500
2540 IF T=5 THEN ANNEAU(2)=1:ECUS=ECUS+500
2550 IF T=6 THEN ANNEAU(3)=1:ECUS=ECUS+500
2560 CONTENU(X,Y)=0
2570 IF ANNEAU(1)=1 AND ANNEAU(2)=1 AND ANNEAU(3)=1 THEN X=14:Y=
6:GOTO 690
2580 GOTO 1580
2590 IF ZY=1 THEN 2600 ELSE GOTO 1590
2600 CLS:BORDER 0
2610 INK 1,24,6:SPEED INK 5,5
2620 PEN 1:PRINT:PRINT:PRINT"          BRAVO!":PRINT:PEN 6
2630 PRINT:PRINT" Valeureux chevalier"
2640 PRINT"vous etes sorti vic-"
2650 PRINT"torieux du terrible"
2660 PRINT:PRINT"Manoir de Rochebrune"
2670 PRINT:PRINT" Et vous avez reuni":PRINT
2680 PRINT"les trois anneaux de"
2690 PRINT"puissance ":PRINT
2700 INK 11,2:PEN 11:PRINT"          ":CHR$(232)::PEN 4:PRINT" ":C
HR$(232)::PEN 6:PRINT" ":CHR$(232)
2710 PEN 7:PRINT"          ":CHR$(234):" ":CHR$(234):" ":CHR$(234)
:END
2720 CLS:BORDER 0
2730 INK 1,4,1:SPEED INK 5,5
2740 PEN 1:PRINT:PRINT:PRINT"          ADIEU!":PRINT:PEN 4
2750 PRINT:PRINT" Valeureux chevalier"
2760 PRINT"les monstres et les":PRINT
2770 PRINT"esprits du Manoir de"
2780 PRINT"Rochebrune ont vain-"
2790 PRINT"cu votre courage et":PRINT
2800 PRINT"votre temerite.":PRINT
2810 PRINT" Vous n'avez pas":PRINT
2820 PRINT"trouve les anneaux":PRINT
2830 PRINT"de puissance : "
2840 GOTO 2700

```

# SQUASH

Je serai bref dans la description de ce programme que vous connaissez certainement : vous devez renvoyer la balle contre les murs à l'aide de votre raquette que vous maniez avec les flèches droite et gauche. Un très bon entraînement pour la précision...

## *QUELQUES DETAILS TECHNIQUES*

70-140	Initialisation
150-160	Affichage du nombre de balles
170-200	Test clavier
210-220	Test de sortie de l'écran
230-260	Affichage de la raquette
270-280	Test de rebond
290-300	Affichage de la balle
310-330	Test de frappe
340-360	Affichage du score
370-400	Fin d'une balle
410-480	Initialisation de l'écran
490-550	Page de présentation
560-670	Fin

## *TABLEAU DES VARIABLES*

X	Coordonnée horizontale de la balle
Y	Coordonnée verticale de la balle

B Nombre de balles  
 V Coordonnée horizontale de la raquette  
 I\$ Scrutation clavier  
 D Déplacement vertical de la balle  
 C Déplacement horizontal de la balle  
 SC Score

```

10 /          *****
20 /          *SQUASH*
30 /          *****
40 /
50 /          par S,L & P Kerloch
60 /
70 ENV 1,15,-1,4
80 X=1:Y=1
90 B=3
100 GOSUB 490
110 BORDER 3
120 /
130 MODE 1:INK 1,18:INK 0,0:INK 2,6:INK 3,14
140 GOSUB 410
150 LOCATE V,24:PRINT"  "
160 LOCATE 1,25:PRINT"BALLES:";B
170 I$=INKEY$
180 IF I$="" THEN 240
190 IF I$=CHR$(242) AND V>0 THEN V=V-2
200 IF I$=CHR$(243) AND V<40 THEN V=V+2
210 IF V<1 THEN V=1
220 IF V>39 THEN V=39
230 PEN 2
240 LOCATE V,24:PRINT CHR$(131):CHR$(131)
250 PEN 3
260 LOCATE X,Y:PRINT"  "
270 Y=Y+D:IF Y>23 OR Y<2 THEN D=-D:SOUND 2,400,60,15,1
280 X=X+C:IF X>39 OR X<2 THEN C=-C:SOUND 2,300,60,15,1
290 PEN 3
300 LOCATE X,Y:PRINT CHR$(231)
310 IF X=V AND Y=23 THEN SC=SC+17:SOUND 2,200,60,15,1:D=-D
  
```

```

320 IF X=V+1 AND Y=23 THEN SC=SC+17:SOUND 2,200,60,15,1:D=-D
330 IF Y=24 THEN GOSUB 370
340 PEN 2
350 LOCATE 18,25:PRINT"SCORE=":SC
360 GOTO 150
370 B=B-1:IF B=0 THEN SOUND 2,0,100,0:SOUND 2,478,60,15,1:SOUND
2,379,60,15,1:SOUND 2,319,60,15,1:SOUND 2,239,60,15,1:GOTO 560
380 IF B>0 THEN LOCATE 1,24:PRINT"
      " / 40espaces
390 X=INT(RND(1)*35)+2:Y=2
400 RETURN
410 '
420 V=15:SC=0:C=1:D=1
430 CLS
440 Z=INT(RND(1)*35)+2:Y=2
450 PEN 1
460 LOCATE 1,25:PRINT"===== " /
39 caracteres
470 PEN 2
480 RETURN
490 MODE 0:INK 1,18,6:SPEED INK 25,25:PEN 1:BORDER 2
500 LOCATE 8,10:PRINT"SQUASH"
510 INK 2,15:PEN 2
520 LOCATE 8,11:PRINT STRING$(6,CHR$(208))
530 FOR A=1 TO 5000
540 NEXT A
550 RETURN
560 MODE 0
570 INK 1,6,2:INK 2,24:INK 3,6:INK 4,15:BORDER 2
580 PEN 1
590 LOCATE 8,5:PRINT"FIN"
600 PEN 4
610 LOCATE 6,6:PRINT STRING$(3,CHR$(208))
620 PEN 2
630 LOCATE 2,15:PRINT"VOTRE SCORE:":
640 PEN 3
650 PRINT SC
660 I$=INKEY$
670 IF I$=" " THEN RUN ELSE 660

```

# LABYRINTHE

Vous êtes égyptologue et vous allez explorer la pyramide de Mykérinos. Mais vous êtes perdu dans le labyrinthe et vous ne disposez que d'une torche pour éclairer autour de vous. Vous utiliserez les quatre flèches pour vous déplacer. En appuyant sur la touche "COPY", vous verrez le plan se dessiner et votre position. Mais cette opération ne se renouvellera que trois fois.

Prenez garde à ne pas vous égarer, on ne viendra pas vous chercher...

## *QUELQUES DETAILS TECHNIQUES*

80-380	Chargement du plan et initialisation
390-520	Affichage de l'explorateur
520-580	Scrutation clavier
590-640	Vous avez gagné !
650-680	Vous avez perdu !
1000-1120	Dessin du labyrinthe
2000-2040	Page de présentation

## *TABLEAU DES VARIABLES*

CASE(40,23)	Labyrinthe
X	Coordonnée horizontale de l'explorateur
Y	Coordonnée verticale de l'explorateur
ENERGIE	Nombre de points d'énergie qu'il vous



```

190 DATA 2,0,1,0,0,0,1,0,0,0,1,0,0,0,1,1,0,1,0,0,1,1,0,1,1,1,0,1
,0,1,0,0,0,1,0,1,0,1,0,0
200 DATA 1,1,1,0,1,0,1,0,1,0,1,0,1,0,1,1,0,1,0,0,0,0,0,1,0,1,0,0
,0,1,1,0,1,1,0,1,0,1,0,1
210 DATA 1,0,0,0,1,0,0,0,1,0,0,0,1,0,0,0,0,1,1,1,1,1,0,1,0,1,0,1
,1,1,0,0,1,0,0,1,0,1,0,1
220 DATA 1,0,1,0,1,1,1,1,1,1,1,1,1,1,1,0,1,1,0,0,0,1,0,0,0,0,0,1
,0,0,0,1,1,0,1,1,0,1,0,1
230 DATA 1,0,0,0,0,0,0,0,0,0,0,0,0,0,0,1,0,1,1,0,0,0,1,1,1,1,1,1,1
,0,0,0,1,1,0,0,1,0,1,0,1
240 DATA 1,0,1,1,1,1,0,1,1,0,1,1,1,0,1,0,1,1,0,1,1,1,1,0,0,0,1,1
,0,1,1,1,1,1,0,1,0,0,0,1
250 DATA 1,0,0,0,1,0,0,0,1,0,1,0,0,0,0,0,0,1,0,0,0,1,0,0,0,0,0,0,0
,0,1,1,0,0,0,0,1,0,0,0,1
260 DATA 1,1,1,0,1,0,1,0,1,0,1,0,1,0,1,0,1,0,1,0,1,0,1,0,0,0,1,1
,1,1,1,0,0,1,0,1,1,1,1,1
270 DATA 1,0,0,0,1,0,0,0,1,0,0,0,0,0,1,0,0,0,0,0,0,1,0,1,1,1,1,1,0
,0,0,0,0,0,1,0,1,0,0,0,1
280 DATA 1,0,0,0,1,0,1,1,1,1,1,1,1,1,1,1,1,1,1,1,0,1,0,0,0,1,0
,1,0,1,0,0,1,1,1,0,1,0,1
290 DATA 1,0,0,0,1,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,1,0,0,0
,0,0,1,0,0,0,0,0,0,0,0,1
300 DATA 1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1
,1,1,1,1,1,1,1,1,1,1,1,1
310 INK 0,0:INK 1,6:INK 2,24:INK 3,1
320 MODE 1
330 WINDOW #0,1,40,2,25:WINDOW #1,1,40,1,1:PAPER 3:CLS
340 DIM CASE(40,23)
350 FOR B=1 TO 23
360 FOR A=1 TO 40
370 READ CASE(A,B)
380 NEXT A,B
390 GOSUB 1000:PAPER 3:CLS
400 PEN 1:PAPER 0:PRINT #1,"          ENERGIE :
"
410 ENERGIE=120
420 CHANCES=3
430 X=2:Y=12

```

```

440 FOR A=X-1 TO X+1
450 FOR B=Y-1 TO Y+1
460 IF CASE(A,B)=0 THEN PAPER 2
470 IF CASE(A,B)=1 THEN PAPER 3
480 IF CASE(A,B)=2 THEN PAPER 2:PEN 1:LOCATE A,B:PRINT CHR$(203)
:GOTO 500
490 LOCATE A,B:PRINT " "
500 NEXT B,A
510 PAPER 2:PEN 0:LOCATE X,Y:PRINT CHR$(249)
520 I$=INKEY$
530 IF I$=CHR$(240) AND CASE(X,Y-1)=0 THEN Y=Y-1:GOTO 590
540 IF I$=CHR$(241) AND CASE(X,Y+1)=0 THEN Y=Y+1:GOTO 590
550 IF I$=CHR$(242) AND CASE(X-1,Y)=0 THEN X=X-1:GOTO 590
560 IF I$=CHR$(243) AND CASE(X+1,Y)=0 THEN X=X+1:GOTO 590
570 IF I$=CHR$(224) AND ENERGIE>10 AND CHANCES>0 THEN GOSUB 1000
:ENERGIE=ENERGIE-10:PAPER 3:CHANCES=CHANCES-1:CLS:GOTO 600
580 GOTO 520
590 PAPER 3:CLS:ENERGIE=ENERGIE-1
600 LOCATE #1,15,1:PRINT #1,ENERGIE
610 IF ENERGIE=0 THEN CLS:PAPER 0:PEN 1:PRINT #1,"          You
s avez perdu !!          " ELSE 650
620 ENV 1,15,-1,9:ENV 2,15,-1,6:ENV 3,15,-1,3::SOUND 2,478,60,15
,1:SOUND 2,478,45,15,2:SOUND 2,478,15,15,3:SOUND 2,478,60,15,1
630 SOUND 2,402,45,15,2:SOUND 2,426,15,15,3:SOUND 2,426,45,15,2:
SOUND 2,478,15,15,3:SOUND 2,478,45,15,2:SOUND 2,506,15,15,3:SOUN
D 2,478,60,15,1
640 FOR a=1 TO 3000:NEXT a:RUN
650 IF X=40 AND Y=12 THEN PAPER 0:PEN 1:PRINT #1,"          ~ You
s avez gagne !!          " ELSE 690
660 ENT 1,5,-1,1,5,1,1,5,-1,1,5,1,1:SOUND 1,0,375,0:SOUND 4,0,37
5,0
670 SOUND 2,478,120,15,0,1:SOUND 2,358,120,15,0,1:SOUND 2,319,15
,15,0,1:SOUND 2,358,15,15,0,1:SOUND 2,319,90,15,0,1:SOUND 2,213,
15,15,0,1:SOUND 2,239,120,15,0,1:SOUND 1,478,120,12,0,1:SOUND 4,
119,120,12,0,1
680 FOR a=1 TO 3000:NEXT a:RUN
690 GOTO 440
1000 FOR B=2 TO 23

```

```
1010 FOR A=1 TO 40
1020 IF CASE(A,B)=0 THEN PAPER 2
1030 IF CASE(A,B)=1 THEN PAPER 3
1040 IF CASE(A,B)=2 THEN PAPER 2:PEN 1:LOCATE A,B:PRINT CHR$(203)
)GOTO 1070
1050 IF A=X AND B=Y THEN 1070
1060 LOCATE A,B:PRINT" "
1070 NEXT A,B
1080 FOR B=1 TO 23
1090 FOR A=1 TO 40
1100 PAPER 3:LOCATE A,B:PRINT" "
1110 NEXT A,B
1120 RETURN
2000 CLS:MODE 0:INK 1,18,6:INK 2,15:BORDER 2:SPEED INK 20,20:PEN
 1:INK 0,0:PAPER 0
2010 LOCATE 6,12:PRINT"LABYRINTHE":PEN 2
2020 FOR A=6 TO 15:LOCATE A,13:PRINT CHR$(208):NEXT A
2030 FOR A=1 TO 2000:NEXT A
2040 BORDER 0:RETURN
```

# PANIC

Cette fois-ci, vous jouerez à deux jeux en même temps. Du côté gauche, vous êtes dans une frêle embarcation, et votre mission est de rattraper les parachutistes imprudents. Du côté droit, vous devez recueillir dans votre seau les gouttes d'eau qui coulent d'une canalisation percée. Au début du jeu, l'AMSTRAD vous demande le niveau de difficulté, il augmente tous les vingts points. Vous perdez des points si une goutte d'eau ou un parachutiste vous échappe. Gardez votre sang froid !

## *QUELQUES DETAILS TECHNIQUES*

70-90	Page de présentation
100-190	Initialisation et redéfinition
200-300	Décor
310-320	Point de départ de la goutte et du parachutiste
330-440	Affichage des gouttes et des parachutistes
450-500	Scrutation clavier
510-550	Calcul du rattrapage d'une goutte ou d'un parachutiste
560-620	Calcul et affichage des points
630-690	Fin

## TABLEAU DES VARIABLES

X	Coordonnée horizontale du bateau
POINT	Score
G\$(50)	Caractère correspondant à une goutte ou un parachutiste
X(50)	Coordonnée horizontale d'une goutte ou d'un parachutiste
Y(50)	Coordonnée verticale d'une goutte ou d'un parachutiste
I(50)	Couleur d'une goutte ou d'un parachutiste
NIVEAU	Niveau de difficulté
RN	Coordonnée horizontale du point de départ d'une goutte ou d'un parachutiste
I\$	Touche frappée au clavier
NG	Nombre de gouttes et de parachutistes

```

10 '          *****
20 '          *PANIC*
30 '          *****
40 '
50 '                par S,L & P Kerloch
60 '
70 MODE 0:INK 1,18,6:INK 2,15:SPEED INK 20,20:BORDER 2:PAPER 0:I
NK 0,0:CALL &BD19
80 PEN 1:LOCATE 8,12:PRINT"PANIC"
90 PEN 2:LOCATE 8,13:PRINT STRING$(5,CHR$(208))
100 X=10:G=1:POINT=5:DIM G$(50),X(50),Y(50),I(50)
110 FOR Z=1 TO 3000:NEXT Z
120 INPUT "NIVEAU (1-5) ":NIVEAU
130 SPEED KEY 2,2
140 SYMBOL AFTER 208
150 SYMBOL 209,0,16,247,247,247,247,16,0
160 SYMBOL 210,24,126,255,66,36,24,24,36
170 SYMBOL 211,0,16,16,58,124,254,254,124
180 SYMBOL 212,0,0,0,0,195,255,126,60
190 SYMBOL 213,14,14,4,236,244,228,10,17

```

```

200 MODE 1:INK 0,0:INK 1,2:INK 2,6:INK 3,20:BORDER 0
210 PEN 2:PRINT"NIVEAU :          POINTS : "
220 PEN 2:PAPER 1
230 FOR I=2 TO 25:LOCATE 20,1:PRINT CHR$(207):NEXT I
240 PEN 3:PAPER 0
250 FOR I=21 TO 40:LOCATE I,2:PRINT CHR$(209):NEXT I
260 PEN 1
270 FOR I=1 TO 19:LOCATE I,25:PRINT CHR$(196):NEXT I
280 PEN 1:PAPER 3:FOR I=20 TO 39:LOCATE I,25:PRINT CHR$(207):NEX
T I
290 PEN 2:PAPER 0
300 LOCATE X,24:PRINT CHR$(212):LOCATE X+20,24:PRINT CHR$(213)

310 RN=INT(RND(1)*38)+1:N=N+1:NG=NG+1
320 IF RN>19 THEN RN=RN+1:G$(G)=CHR$(211):X(G)=RN:Y(G)=3:I(G)=1
ELSE G$(G)=CHR$(210):X(G)=RN:Y(G)=2:I(G)=3
330 FOR G=1 TO NIVEAU
340 IF X(G)=0 THEN 360
350 LOCATE X(G),Y(G):PRINT " "
360 NEXT G
370 FOR G=1 TO NIVEAU
380 IF X(G)=0 THEN 430
390 Y(G)=Y(G)+1
400 PEN I(G):LOCATE X(G),Y(G):PRINT G$(G)
410 IF Y(G)=24 AND X(G)<20 THEN W=1:G=G-1:LOCATE X(G),24:PRINT"
":POINT=POINT-1:NG=NG-1:GOTO 600
420 IF Y(G)=24 AND X(G)>20 THEN W=1:G=G-1:LOCATE X(G),24:PRINT"
":POINT=POINT-1:NG=NG-1:GOTO 600
430 NEXT G
440 IF POINT=0 THEN 630
450 I$=INKEY$:PEN 2:LOCATE X,24:PRINT CHR$(212):LOCATE X+20,24:P
RINT CHR$(213)
460 IF I$<>CHR$(242) AND I$<>CHR$(243) THEN 510
470 LOCATE X,24:PRINT" ":LOCATE X+20,24:PRINT" "
480 IF I$=CHR$(242) AND X>1 THEN X=X-1
490 IF I$=CHR$(243) AND X<19 THEN X=X+1
500 PEN 2:LOCATE X,24:PRINT CHR$(212):LOCATE X+20,24:PRINT CHR$(
213)
510 FOR Q=1 TO NIVEAU

```

```

520 IF X(Q)=0 THEN 550
530 IF X(Q)<20 AND Y(Q)=23 AND X(Q)=X THEN POINT=POINT+1:G=G-1:N
G=NG-1:LOCATE X(Q),23:PRINT" ":W=1:GOTO 570
540 IF X(Q)>20 AND Y(Q)=23 AND X(Q)=X+20 THEN POINT=POINT+1:G=G-
1:NG=NG-1:LOCATE X(Q),23:PRINT" ":W=1:GOTO 570
550 NEXT Q
560 W=INT(RND(1)*6)+1
570 PEN 2:LOCATE 29,1:PRINT POINT
580 IF POINT/20=INT(POINT/20) THEN POINT=POINT+5:NIVEAU=NIVEAU+1
:G=1:FOR A=1 TO 50:G$(A)="":X(A)=0:Y(A)=0:I(A)=0:NEXT A:NG=0:N=0
:GOTO 200
590 PEN 2:LOCATE 9,1:PRINT NIVEAU
600 IF W=1 AND NG<NIVEAU THEN G=G+1:GOTO 310
610 IF NG=0 THEN 310
620 GOTO 330
630 MODE 0:INK 1,18,6:INK 2,15:SPEED INK 20,20:BORDER 2:PAPER 0:
INK 0,0:SPEED KEY 20,2
640 PEN 1:LOCATE 9,10:PRINT"FIN"
650 PEN 2:LOCATE 9,11:PRINT STRING$(3,CHR$(208))
660 PRINT:INPUT"VOULEZ-VOUS REJOUER ":REP$
670 IF REP$="O" OR REP$="o" THEN RUN
680 IF REP$="N" OR REP$="n" THEN END
690 GOTO 660

```

# HISTOGRAMME

Cet utilitaire se présente sous la forme d'un sous-programme que vous pourrez insérer dans vos propres programmes de gestion ou de statistiques.

Pour cela, vous devrez modifier une partie du programme (lignes 10 à 60) pour remplir la fiche H(MOIS) avec les résultats provenant de votre programme. Vous pouvez aussi remarquer que cette routine ne comprend ni GOTO ni GOSUB pour faciliter son insertion. Vous pourrez modifier les valeurs DISTANCE, MARGE et LARGEUR à votre gré.

## *QUELQUES DETAILS TECHNIQUES*

10-60	Entrée des données (remplissage du fichier H(MOIS)).
70-100	Initialisation
110-120	Tracé de l'axe horizontal
130	Tracé de l'axe vertical
140-190	Graduation de l'axe vertical
210-280	Tracé des deux faces de la barre
290-340	Tracé du dessus de la barre

## *TABLEAU DES VARIABLES*

B	Indicateur de négativité
MOIS	Unité de temps (représente les jours, les mois ou les années)

DISTANCE	Distance entre les barres
MARGE	Distance entre le bord gauche de l'écran et l'axe vertical
LARGEUR	Largeur des barres
H(X)	Tableau des données à visualiser

```

10 '-----ENTREE DES DONNEES-----
20 INPUT "nombre de barres",MOIS
30 DIM H(MOIS+1):B=1
40 FOR A= 1 TO MOIS
50 INPUT H(A)
60 NEXT
70 '-----AXES-----
80 INK 1,2:INK 2,1:INK 3,11:MODE 1:INK 0,0:PAPER 0:BORDER 0
90 DISTANCE=30:MARGE=5:LARGEUR=20
100 CLS:ORIGIN 1,200
110 PLOT MARGE,LARGEUR/4
120 DRAWR 640,0,1
130 PLOT MARGE,200:DRAWR 0,-400,1
140 FOR C=-200 TO 200 STEP 25
150 PLOT MARGE-2,C+LARGEUR/4,1
160 NEXT C
170 FOR C=-200 TO 200 STEP 100
180 PLOT MARGE-4,C+LARGEUR/4,1
190 NEXT C
200 '-----TRACE-----
210 FOR I=1 TO MOIS
220 IF H(I)=0 THEN H(I)=1
230 FOR J=1 TO H(I) STEP 2*SGN(H(I))
240 PLOT I*DISTANCE+MARGE,J,1
250 DRAWR LARGEUR,0,1
260 PLOT I*DISTANCE+MARGE+LARGEUR,J,2
270 DRAWR LARGEUR/2,LARGEUR/2,2
280 NEXT J
290 IF H(I)<0 THEN B=0 ELSE B=1
300 FOR R=0 TO LARGEUR/2 STEP 2
310 PLOT I*DISTANCE+MARGE+R,J*B+R,3
320 DRAWR LARGEUR,0,3
330 NEXT R
340 NEXT I

```

# CAMEMBERT

Ce n'est pas d'un fromage qu'il s'agit, mais plutôt d'un sous-programme à insérer dans vos propres programmes de gestion, comme l'utilitaire précédent. Il n'y a ni GOSUB, ni GOTO pour faciliter son insertion. Vous pourrez modifier la dimension du tableau de la ligne 10, mais il ne pourra pas dépasser le chiffre 50.

## *QUELQUES DETAILS TECHNIQUES*

10-50	Remplissage du fichier
100-150	Initialisation des couleurs, variables et du mode de définition
160-310	Traçage du camembert
320-380	Traçage de la séparation entre les parts

## *TABLEAU DES VARIABLES*

V(50)	Valeur des 50 parts différentes
r	Rayon
n	Canal de couleur
h	Mode de définition
nbpart	Nombre de parts
f	Valeur des parts
p	Pas du cercle
x	Coordonnée horizontale du cercle
y	Coordonnée verticale du cercle
k	Couleur de la bordure du cercle

```

10 DIM v(50)
20 INPUT "nombre de parts:":nbpart
30 FOR e=1 TO nbpart
40 INPUT "valeur:":v(e)
50 NEXT e
100 b=3.6:r=150:p=0.7:n=1
110 FOR l=0 TO 15:READ m:INK l,m:NEXT
120 DATA 0,18,2,6,24,7,21,13,5,10,15,17,19,3,8,12
130 INPUT "mode de definition (0/1/2)":h
140 MODE h
150 DEG:ORIGIN 320,200
160 FOR j=1 TO nbpart
170 f=v(j)
180 FOR i=d*b TO (f+d)*b STEP p
190 x=r*COS(i)
200 y=r*SIN(i)
210 PLOT 0,0
220 IF h=2 AND n=2 THEN n=0:k=1
230 IF h=1 AND n=4 THEN n=0:k=1
240 IF h=0 AND n=16 THEN n=0:k=1
250 DRAW x,y,n
260 PLOT x,y,k
270 NEXT i
280 k=0
290 d=f+d
300 n=n+1
310 NEXT j
320 FOR a=0 TO nbpart
330 x=r*COS(c*b)
340 y=r*SIN(c*b)
350 PLOT 0,0
360 DRAW x,y,0
370 c=c+v(a)
380 NEXT a

```

# FONCTION

Ce programme, de fonctionnement simple, s'adresse aussi bien aux mathématiciens qu'aux lycéens. Il représente, sous forme graphique, l'évolution d'une fonction mathématique. Au début, l'AMSTRAD vous demande l'intervalle d'étude, la précision de la courbe, la grandeur de la courbe horizontalement et verticalement. Il vous demande aussi si vous voulez effacer la courbe précédente ou la comparer avec la suivante.

## *QUELQUES DETAILS TECHNIQUES*

10-110	Initialisation et affichage du menu
120-170	Entrée des paramètres
180-260	Traçage et calcul de la fonction
270-320	Fin du traçage
1000-1120	Traçage des axes
2000-2070	Affichage des points de la fonction

## *TABLEAU DES VARIABLES*

A	Intervalle minimum
B	Intervalle maximum
P	Précision
C	Echelle horizontale
E	Echelle verticale
X	Coordonnée horizontale de la fonction

Y Coordonnée verticale de la fonction  
A\$ Test clavier à la fin  
I Coordonnée horizontale du point  
J Coordonnée verticale du point

```
1 '          *****  
2 '          *FONCTION*  
3 '          *****  
4 '  
5 '                par S,L & P Kerloch  
6 '  
10 MODE 1  
20 INK 0,0:INK 1,9:INK 2,6:INK 3,15  
30 PRINT: PRINT" COMMANDES :"  
40 PRINT"  A= ARRETER"  
50 PRINT"  C= MODIFIER LA FONCTION"  
60 PRINT"  R= RECOMMENCER"  
70 PRINT  
80 WINDOW #1,1,40,25,25  
90 PAPER #1,0:PEN #1,3  
100 CLS #1  
110 ORIGIN 320,200  
120 INPUT #1, "intervale   :":A,B  
130 INPUT #1,"precision   :":P  
140 INPUT #1,"echelle     :":C,E  
150 INPUT #1,"effacement  :":B$  
160 PRINT#1  
170 IF B$="0" OR B$="o" THEN CLS  
180 GOSUB 1000 ' AFF AXES  
190 X=A  
200 IF X=0 THEN X=10-8  
210 Y=SIN(X)  
220 LOCATE 1,1:PRINT"X=":ROUND(X,2)  
230 LOCATE 32,1:PRINT"Y=":ROUND(Y,2)  
240 GOSUB 2000 ' AFF POINT  
250 X=X+P
```

```
260 IF X<B THEN 200
270 PRINT #1,"FIN"
280 A$=INKEY$
290 IF A$="A" OR A$="a" THEN END
300 IF A$="R" OR A$="r" THEN 120
310 IF A$="C" OR A$="c" THEN LIST 210
320 GOTO 280
1000 PLOT -320,-185,1
1010 DRAWR 640,0,1
1020 PLOT 0,199,1
1030 DRAW 0,-185,1
1040 PLOT -320,0,1
1050 DRAW 640,0,1
1060 FOR F=-180 TO 200 STEP 20
1070 PLOT 2,F,1
1080 NEXT
1090 FOR G=-320 TO 320 STEP 20
1100 PLOT G,-2,1
1110 NEXT
1120 RETURN
2000 I=C*20*X
2010 J=20*E*Y
2020 IF I>320 THEN I=320
2030 IF I<-320 THEN I=-320
2040 IF J>199 THEN J=199
2050 IF J<-184 THEN J=-184
2060 PLOT I,J,2
2070 RETURN
```



# LES TOURS DE HANOÏ

Il s'agit d'un casse-tête chinois. Au début du jeu, vous choisissez le nombre d'étages que possédera votre tour. Elle s'affichera en entier sur la partie gauche de l'écran. Vous devrez la déplacer en entier sur la partie de droite, et ceci étage par étage, en un minimum de coups. Les étages s'empilent du plus grand au plus petit, comme une pyramide. Une clochette retentit quand vous avez gagné. Bonne chance !

## *QUELQUES DETAILS TECHNIQUES*

70-210	Initialisation
220-460	Votre jeu
470-570	Affichage des tours
600-650	Page de présentation

## *TABLEAU DES VARIABLES*

ND	Nombre d'étages
A(3,11)	Etages des tours
N	Tour
I	Etage
NM	Nombre de tentatives
K	Tour de départ et d'arrivée

```

10 ' *****
20 ' *LES TOURS DE HANOI*
30 ' *****
40 '
50 '                par S,L & P Kerloch
60 '
70 GOSUB 580
80 MODE 2:BORDER 0:INK 0,0:PAPER 0:PEN 1:INK 1,15
90 CLS:PRINT
100 WINDOW #0,1,80,9,24
110 CN=1
120 DIM A(3,11)
130 INPUT "Nombre d'etages ( 1 a 10 ) ":ND
140 IF ND<1 OR ND>10 THEN 130
150 CLS:WINDOW #2,1,80,25,25:PAPER #2,1:PEN #2,0
160 FOR I=11-ND TO 10
170 A(1,I)=I
180 NEXT I
190 LOCATE #3,32,4:PRINT #3,"LES TOURS DE HANOI"-
200 LOCATE #3,32,5:PRINT #3,STRING$(18,CHR$(208))
210 LOCATE #3,1,8:PRINT #3,"          1
          3"
220 CLS #0:GOSUB 470
230 FOR I=11-ND TO 10
240 IF A(3,I)<>I THEN 270
250 NEXT I
260 PRINT #2,"Felicitations ! Vous avez reussi en";NM:"deplaceme
nts":ENV 1,15,-1,4:SOUND 2,50,60,15,1:END
270 NM=NM+1
280 INPUT #2,"De quelle tour voulez-vous oter un anneau ":K

```

2

```

290 IF K<1 OR K>3 THEN 280
300 FOR I=1 TO 10
310 IF A(K,I)=0 THEN 350
320 TH=A(K,I)
330 A(K,I)=0
340 GOTO 370
350 NEXT I
360 GOTO 280
370 INPUT #2,"Vers quelle tour voulez-vous deplacer cet anneau "
:K
380 IF K<1 OR K>3 THEN 370
390 FOR I=1 TO 10
400 IF A(K,I)=0 THEN 440
410 IF A(K,I)<TH THEN PRINT #2,"Vous ne pouvez pas le faire ":GO
TO 370
420 A(K,I-1)=TH
430 GOTO 210
440 NEXT I
450 A(K,10)=TH
460 GOTO 210
470 FOR N=1 TO 3
480 FOR I=10 TO 11-ND STEP -1
490 X=N*26-25
500 IF A(N,I)=0 THEN 540
510 X=X+10-A(N,I)
520 LOCATE X,I+4:PRINT CHR$(143):
530 FOR J=2 TO (A(N,I)*2)-1:PRINT CHR$(143)::NEXT J:PRINT""
540 NEXT I
550 NEXT N
560 LOCATE 1,25
570 RETURN
580 MODE 0:INK 1,18,6:INK 2,15:INK 0,0:PEN 1:PAPER 0:BORDER 2:SP
EED INK 20,20
585 PEN 1:LOCATE 2,12:PRINT"LES TOURS DE HANOI"
590 PEN 2
600 LOCATE 1,13:PRINT " ":STRING$(18,CHR$(208))
610 FOR A=1 TO 4000:NEXT A
620 RETURN

```



## PUISSANCE 4

Vous connaissez sûrement ce très célèbre jeu. Nous vous proposons d'y jouer, mais cette fois-ci contre votre ordinateur. Au début, une grille de huit colonnes se dessine sur l'écran. Pour gagner, il suffit d'aligner 4 pions horizontalement ou diagonalement. Mais votre AMSTRAD fait tout pour vous en empêcher. Vous avez la couleur jaune.

Encore un jeu plein de malice, qui vous donnera du fil à retordre.

### *QUELQUES DETAILS TECHNIQUES*

10-990	Tests
1230-1330	Affichage de la grille
1340-1380	Affichage des points rouges et jaunes
1390-1450	Page de présentation

```

2 '      *****
3 '      *PUISSANCE 4*
4 '      *****
5 '
6 '      par S,L & P Kerloch
7 '
10 GOSUB 1390
20 MODE 1:INK 1,2:INK 2,6:INK 3,24:BORDER 0
30 PRINT CHR$(22)+CHR$(1)
40 WINDOW #2,1,40,24,25
50 DIM B$(8,8),L(8),S(4),F(4),V(16),N(4)
60 DATA 1,100,500,1E20,1,800,4000,1E20,1,75,900,1E18,1,450,3000,
1E18
70 FOR I=1 TO 16:READ V(I):NEXT I
80 X$=CHR$(230):O$=CHR$(231)
90 FOR I=1 TO 8:FOR J=1 TO 8:B$(I,J)=" ":NEXT J,I
100 FOR I=1 TO 8:L(I)=0:NEXT I
110 GOSUB 1230
120 GOTO 140
130 PRINT #2,"coup impossible !!      recommencez"
140 PRINT #2,"a vous de jouer ..."
150 SD=2
160 INPUT #2,M
170 IF M>8 OR M<1 THEN 160
180 CO=22
190 L=L(M)
200 IF L>7 THEN 130
210 L(M)=L+1:L=L+1
220 B$(L,M)=X$
230 GOSUB 1340
240 PRINT #2:PRINT #2,"a moi de jouer ..."
250 P$=X$
260 GOSUB 880
270 FOR Z=1 TO 4
280 IF S(Z)<4 THEN 310
290 PEN 3:LOCATE 12,12:PRINT"vous avez gagne !!"
300 PRINT CHR$(22)+CHR$(0):END
310 NEXT Z
320 M9=0:V1=0

```

```

330 SD=1
340 N1=1
350 FOR U=1 TO 8
360 L=L(U)+1
370 IF L>8 THEN 720
380 V=1
390 P$=0$:W=0
400 M=U
410 GOSUB 880
420 FOR I=1 TO 4:N(I)=0:NEXT I
430 FOR Z=1 TO 4
440 S=S(Z)
450 IF S-W>3 THEN 770
460 T=S+F(Z)
470 IF T<4 THEN 500
480 V=V+4
490 N(S)=N(S)+1
500 NEXT Z
510 FOR I=1 TO 4
520 N=N(I)-1
530 IF N=-1 THEN 560
540 I1=8*W+4*SGN(N)+I
550 V=V+V(I1)+N*V(8*W+I)
560 NEXT I
570 IF W=1 THEN 600
580 W=1:P$=X$
590 GOTO 410
600 L=L+1
610 IF L>8 THEN 660
620 GOSUB 880
630 FOR Z=1 TO 4
640 IF S(Z)>3 THEN V=2
650 NEXT Z
660 IF V<V1 THEN 720
670 IF V>V1 THEN N1=1:GOTO 700
680 N1=N1+1
690 IF RND(1)>1/N1 THEN 720
700 V1=V

```

```

710 M9=U
720 NEXT U
730 IF M9<>0 THEN 760
740 PEN 2:LOCATE 14,12:PRINT"partie nulle":PRINT CHR$(22)+CHR$(0
):END
750 GOTO 1220
760 M=M9
770 '
780 L=L(M)+1:L(M)=L(M)+1
790 E$(L,M)=0$
800 P$=0$:GOSUB 1340
810 GOSUB 880
820 FOR Z=1 TO 4
830 IF S(Z)<4 THEN 860
840 PEN 2:LOCATE 13,12:PRINT"J'ai gagne !!!":PRINT CHR$(22)+CHR$
(0):END
850 GOTO 1220
860 NEXT Z
870 GOTO 140
880 Q$=X$
890 IF P$=X$ THEN Q$=0$
900 D2=1:D1=0
910 Z=0
920 GOSUB 1000
930 D1=1:D2=1
940 GOSUB 1000
950 D2=0:D1=1
960 GOSUB 1000
970 D2=-1:D1=1
980 GOSUB 1000
990 RETURN
1000 D=1:S=1
1010 T=0
1020 Z=Z+1
1030 C=0
1040 FOR K=1 TO 3
1050 M5=M+K*D1:L1=L+K*D2
1060 IF M5<1 OR L1<1 OR M5>8 OR L1>8 THEN 1150
1070 G$=B$(L1,M5)

```

```

1080 IF C=0 THEN 1120
1090 IF G#=Q# THEN K=3:GOTO 1150
1100 T=T+1
1110 GOTO 1150
1120 IF G#=P# THEN S=S+1:GOTO 1150
1130 G=1
1140 GOTO 1090
1150 NEXT K
1160 IF D=0 THEN 1190
1170 D=0:D1=-1:D2=-D2
1180 GOTO 1030
1190 S(2)=S
1200 F(2)=T
1210 RETURN
1220 END
1230 ' AFFICHAGE GRILLE
1240 PEN 2
1250 FOR I=1 TO 8:LOCATE I*2+11,17:PRINT I:NEXT I
1260 PLOT 202,399:DRAW 456,399,1:DRAWR 0,-250,1:DRAW 202,149,1:D
RAWR 0,250,1
1270 FOR A=232 TO 440 STEP 32
1280 PLOT A,399,1:DRAWR 0,-250,1
1290 NEXT A
1300 FOR A=373 TO 173 STEP -32
1310 PLOT 202,A,1:DRAW 456,A,1
1320 NEXT A
1330 RETURN
1340 ' AFFICHAGE PIONS
1350 LOCATE M*2+12,17-L(M)*2
1360 IF B$(L,M)=CHR$(231) THEN PEN 2:PRINT CHR$(231)
1370 IF B$(L,M)=CHR$(230) THEN PEN 3:PRINT CHR$(231)
1380 RETURN
1390 MODE 0:INK 1,18,6:INK 2,15:INK 0,0:BORDER 2:SPEED INK 20,20
1400 PEN 1
1410 LOCATE 5,12:PRINT"PUISSANCE 4"
1420 PEN 2
1430 LOCATE 5,13:PRINT STRING$(11,CHR$(208))
1440 FOR A=1 TO 4000:NEXT A
1450 RETURN

```

## ANNEXE 1

Voici quelques courts programmes modulant des bruits différents que vous pourrez insérer dans vos propres programmes :

### 1) Laser War

```
10 C=INT(RND(1)*5)+1
20 FOR A=1 TO 31
30 SOUND 2,0,C,15,0,0,A
40 NEXT A
50 GOTO 10
```

### 2) Soucoupe volante

```
10 ENT 1,10,1,1,10,-1,1
20 SOUND 2,400,20,15,0,1
30 GOTO 10
```

### 3) Police

```
10 FOR A=200 TO 100 STEP -1
20 SOUND 2,A,1,15
30 NEXT A
40 FOR A=100 TO 200
50 SOUND 2,A,1,15
60 NEXT A
70 GOTO 10
```

#### 4) Bourdon

```
10 ENV 1,1,-15,1
20 SOUND 2,0,1,15
30 GOTO 10
```

#### 5) Pompiers

```
5 SOUND 1,4,120,15
10 SOUND 2,239,60,15
20 SOUND 2,284,60,15
30 GOTO 10
```

#### 6) Ressort métallique

```
10 ENT 1,10,-1,4
20 ENV 1,15,-1,4
30 SOUND 2,478,60,15,1,1
40 GOTO 10
```

#### 7) Tous aux abris

```
5 ENV 1,1,-1,1,1,1,1
10 FOR A=600 TO 300 STEP -1
20 SOUND 2,A,3,15,1
30 NEXT A
35 FOR A=1 TO 200
40 SOUND 2,300,2,15,1
50 NEXT A
```

#### 8) Ambulance

```
10 SOUND 2,239,20,15
20 SOUND 2,284,20,15
30 SOUND 2,239,20,15
40 SOUND 2,0,20,15
50 GOTO 10
```

#### 9) Corde

```
10 ENV 1,15,-1,6
20 SOUND 2,400,90,15,1
```

### 10) Ressort élastique

```
10 ENV 1,15,-1,4  
20 SOUND 2,0,60,15,1  
30 GOTO 10
```

### 11) Tir laser

```
10 FOR A=1 TO 31  
20 SOUND 2,10,2,13,0,0,A  
30 NEXT A
```

### 12) Explosion

```
10 ENV 1,15,-1,12  
20 SOUND 6,4,180,15,1,0,31
```

### 13) Coup de feu

```
10 ENV 1,15,-1,4  
20 SOUND 6,4,60,15,1,0,10
```

### 14) Moteur d'avion

```
10 ENV 1,1,-3,1,1,3,1  
20 ENV 2,1,5,1,1,-5,1  
30 SOUND 1,1000,10,15,0,1,31  
40 SOUND 2,300,10,15,0,1,1  
50 SOUND 4,3000,10,15,0,2,10  
60 GOTO 10
```

### 15) Mitraillette

```
10 ENV 1,15,-1,2  
20 SOUND 2,0,10,15,1,0,10  
30 GOTO 10
```

### 16) Ebullition

```
10 A=INT(RND(1)*400)+50  
20 SOUND 2,A,1,15  
30 GOTO 10
```

### 17) Alerte

```
5 ENT 1,10,1,4
10 ENV 1,3,-1,1
15 SOUND 2,500,4,15,1,1
20 FOR A=1 TO 20
30 SOUND 2,500,4,15,1
40 NEXT A
50 SOUND 2,0,5,15
60 GOTO 10
```

### 18) Moteur

```
10 ENT 1,10,-1,1,10,1,1
20 SOUND 2,0,20,15,0,1
30 GOTO 10
```

### 19) Ultrason

```
10 SOUND 2,6,10,15
20 GOTO 10
30 GOTO 10
```

### 20) Galop

```
10 ENV 1,13,-1,2
20 SOUND 2,0,15,15,1,0,10
30 SOUND 2,0,15,15,1,0,10
40 SOUND 2,0,30,15,1,0,20
50 GOTO 10
```

## ANNEXE 2

Vous avez sans doute remarqué que l'AMSTRAD ne possède pas la fonction bien pratique de dessiner un cercle. Maintenant, grâce à ce court programme, c'est chose faite.

Quand A varie de 0 à 360°, le sinus représente la coordonnée verticale du cercle, et le cosinus représente la coordonnée horizontale.

R = Rayon

U = Coordonnée horizontale du centre

V = Coordonnée verticale du centre

```
10 DEG
15 PLOT U,V
20 FOR A=0 TO 1170
30 X=R*COS(A)+U
40 Y=R*SIN(A)+V
50 PLOT X,Y,1
60 NEXT A
```

# SOMMAIRE

Préface	7
---------	---

## Programmes graphiques

— Démonstration couleur	9
— Papillons	13
— Télécran	17
— Courbes folles	21

## Programmes musicaux

— Tendre mélodie	25
— Boîte à rythmes	29
— Clavier	31

## Jeux

— Bombardier	35
— Pac Man	43
— 3D Driver	39
— Steeple-Chase	49
— Le manoir de Rochebrune	55
— Squash	67
— Labyrinthe	69
— Panic	75

## **Programmes utilitaires**

- Histogramme \_\_\_\_\_ 79
- Camembert \_\_\_\_\_ 81
- Fonction \_\_\_\_\_ 83

## **Programmes ludiques**

- Les tours de Hanoï \_\_\_\_\_ 87
- Puissance 4 \_\_\_\_\_ 91

## **Annexe 1**

Différents sons et bruits à insérer dans des programmes personnels. \_\_\_\_\_ 96


## **Annexe 2**

Comment dessiner un cercle. \_\_\_\_\_ 100









**PRIX 45 F**

RESEARCHERS AT THE UNIVERSITY OF MICHIGAN




Document **numérisé**  
avec amour par :

**AMSTRAD**

CPC 

MÉMOIRE ÉCRITE



<https://acpc.me/>